
1 Introduction

Our purpose in writing this tutorial is not to teach programming, nor even to teach functional
programming. Rather, it is intended to serve as a supplement to the Haskell Report [4], which is
otherwise a rather dense technical exposition. Our goal is to provide a gentle introduction to Haskell
for someone who has experience with at least one other language, preferably a functional language
(even if only an \almost-functional" language such as ML or Scheme). If the reader wishes to learn
more about the functional programming style, we highly recommend Bird's text Introduction to
Functional Programming [1] or Davie's An Introduction to Functional Programming Systems Using
Haskell [2]. For a useful survey of functional programming languages and techniques, including
some of the language design principles used in Haskell, see [3].

The Haskell language has evolved signi�cantly since its birth in 1987. This tutorial deals with
Haskell 98. Older versions of the language are now obsolete; Haskell users are encouraged to use
Haskell 98. There are also many extensions to Haskell 98 that have been widely implemented.
These are not yet a formal part of the Haskell language and are not covered in this tutorial.

Our general strategy for introducing language features is this: motivate the idea, de�ne some
terms, give some examples, and then point to the Report for details. We suggest, however, that the
reader completely ignore the details until the Gentle Introduction has been completely read. On the

2 2 VALUES, TYPES, AND OTHER GOODIES

other hand, Haskell's Standard Prelude (in Appendix A of the Report and the standard libraries
(found in the Library Report [5]) contain lots of useful examples of Haskell code; we encourage a
thorough reading once this tutorial is completed. This will not only give the reader a feel for what
real Haskell code looks like, but will also familiarize her with Haskell's standard set of prede�ned
functions and types.

Finally, the Haskell web site, http://haskell.org, has a wealth of information about the
Haskell language and its implementations.

[We have also taken the course of not laying out a plethora of lexical syntax rules at the outset.
Rather, we introduce them incrementally as our examples demand, and enclose them in brackets,
as with this paragraph. This is in stark contrast to the organization of the Report, although the
Report remains the authoritative source for details (references such as \x2.1" refer to sections in
the Report).]

Haskell is a typeful programming language:1 types are pervasive, and the newcomer is best o�
becoming well aware of the full power and complexity of Haskell's type system from the outset. For
those whose only experience is with relatively \untypeful" languages such as Perl, Tcl, or Scheme,
this may be a diÆcult adjustment; for those familiar with Java, C, Modula, or even ML, the
adjustment should be easier but still not insigni�cant, since Haskell's type system is di�erent and
somewhat richer than most. In any case, \typeful programming" is part of the Haskell programming
experience, and cannot be avoided.

2 Values, Types, and Other Goodies

Because Haskell is a purely functional language, all computations are done via the evaluation of
expressions (syntactic terms) to yield values (abstract entities that we regard as answers). Every
value has an associated type. (Intuitively, we can think of types as sets of values.) Examples
of expressions include atomic values such as the integer 5, the character 'a', and the function
\x -> x+1, as well as structured values such as the list [1,2,3] and the pair ('b',4).

Just as expressions denote values, type expressions are syntactic terms that denote type values
(or just types). Examples of type expressions include the atomic types Integer (in�nite-precision
integers), Char (characters), Integer->Integer (functions mapping Integer to Integer), as well
as the structured types [Integer] (homogeneous lists of integers) and (Char,Integer) (character,
integer pairs).

All Haskell values are \�rst-class"|they may be passed as arguments to functions, returned as
results, placed in data structures, etc. Haskell types, on the other hand, are not �rst-class. Types
in a sense describe values, and the association of a value with its type is called a typing. Using the
examples of values and types above, we write typings as follows:

5 :: Integer

'a' :: Char

inc :: Integer -> Integer

[1,2,3] :: [Integer]

('b',4) :: (Char,Integer)
1Coined by Luca Cardelli.



2.1 Polymorphic Types 3

The \::" can be read \has type."

Functions in Haskell are normally de�ned by a series of equations. For example, the function
inc can be de�ned by the single equation:

inc n = n+1

An equation is an example of a declaration. Another kind of declaration is a type signature decla-
ration (x4.4.1), with which we can declare an explicit typing for inc:

inc :: Integer -> Integer

We will have much more to say about function de�nitions in Section 3.

For pedagogical purposes, when we wish to indicate that an expression e1 evaluates, or \re-
duces," to another expression or value e2, we will write:

e1 ) e2

For example, note that:

inc (inc 3) ) 5

Haskell's static type system de�nes the formal relationship between types and values (x4.1.3).
The static type system ensures that Haskell programs are type safe; that is, that the programmer has
not mismatched types in some way. For example, we cannot generally add together two characters,
so the expression 'a'+'b' is ill-typed. The main advantage of statically typed languages is well-
known: All type errors are detected at compile-time. Not all errors are caught by the type system;
an expression such as 1/0 is typable but its evaluation will result in an error at execution time.
Still, the type system �nds many program errors at compile time, aids the user in reasoning about
programs, and also permits a compiler to generate more eÆcient code (for example, no run-time
type tags or tests are required).

The type system also ensures that user-supplied type signatures are correct. In fact, Haskell's
type system is powerful enough to allow us to avoid writing any type signatures at all;2 we say
that the type system infers the correct types for us. Nevertheless, judicious placement of type
signatures such as that we gave for inc is a good idea, since type signatures are a very e�ective
form of documentation and help bring programming errors to light.

[The reader will note that we have capitalized identi�ers that denote speci�c types, such as
Integer and Char, but not identi�ers that denote values, such as inc. This is not just a convention:
it is enforced by Haskell's lexical syntax. In fact, the case of the other characters matters, too: foo,
fOo, and fOO are all distinct identi�ers.]

2.1 Polymorphic Types

Haskell also incorporates polymorphic types|types that are universally quanti�ed in some way
over all types. Polymorphic type expressions essentially describe families of types. For example,
(8a)[a] is the family of types consisting of, for every type a, the type of lists of a. Lists of

2With a few exceptions to be described later.

integers (e.g. [1,2,3]), lists of characters (['a','b','c']), even lists of lists of integers, etc., are
all members of this family. (Note, however, that [2,'b'] is not a valid example, since there is no
single type that contains both 2 and 'b'.)

[Identi�ers such as a above are called type variables, and are uncapitalized to distinguish them
from speci�c types such as Int. Furthermore, since Haskell has only universally quanti�ed types,
there is no need to explicitly write out the symbol for universal quanti�cation, and thus we sim-
ply write [a] in the example above. In other words, all type variables are implicitly universally
quanti�ed.]

Lists are a commonly used data structure in functional languages, and are a good vehicle for
explaining the principles of polymorphism. The list [1,2,3] in Haskell is actually shorthand for
the list 1:(2:(3:[])), where [] is the empty list and : is the in�x operator that adds its �rst
argument to the front of its second argument (a list).3 Since : is right associative, we can also
write this list as 1:2:3:[].

As an example of a user-de�ned function that operates on lists, consider the problem of counting
the number of elements in a list:

length :: [a] -> Integer

length [] = 0

length (x:xs) = 1 + length xs

This de�nition is almost self-explanatory. We can read the equations as saying: \The length of the
empty list is 0, and the length of a list whose �rst element is x and remainder is xs is 1 plus the
length of xs." (Note the naming convention used here; xs is the plural of x, and should be read
that way.)

Although intuitive, this example highlights an important aspect of Haskell that is yet to be
explained: pattern matching. The left-hand sides of the equations contain patterns such as [] and
x:xs. In a function application these patterns are matched against actual parameters in a fairly
intuitive way ([] only matches the empty list, and x:xs will successfully match any list with at least
one element, binding x to the �rst element and xs to the rest of the list). If the match succeeds,
the right-hand side is evaluated and returned as the result of the application. If it fails, the next
equation is tried, and if all equations fail, an error results.

De�ning functions by pattern matching is quite common in Haskell, and the user should become
familiar with the various kinds of patterns that are allowed; we will return to this issue in Section 4.

The length function is also an example of a polymorphic function. It can be applied to a list
containing elements of any type, for example [Integer], [Char], or [[Integer]].

length [1,2,3] ) 3
length ['a','b','c'] ) 3
length [[1],[2],[3]] ) 3

Here are two other useful polymorphic functions on lists that will be used later. Function head

returns the �rst element of a list, function tail returns all but the �rst.

3: and [] are like Lisp's cons and nil, respectively.

head :: [a] -> a

head (x:xs) = x

tail :: [a] -> [a]

tail (x:xs) = xs

Unlike length, these functions are not de�ned for all possible values of their argument. A runtime
error occurs when these functions are applied to an empty list.

With polymorphic types, we �nd that some types are in a sense strictly more general than
others in the sense that the set of values they de�ne is larger. For example, the type [a] is more
general than [Char]. In other words, the latter type can be derived from the former by a suitable
substitution for a. With regard to this generalization ordering, Haskell's type system possesses two
important properties: First, every well-typed expression is guaranteed to have a unique principal
type (explained below), and second, the principal type can be inferred automatically (x4.1.3). In
comparison to a monomorphically typed language such as C, the reader will �nd that polymorphism
improves expressiveness, and type inference lessens the burden of types on the programmer.

An expression's or function's principal type is the least general type that, intuitively, \contains
all instances of the expression". For example, the principal type of head is [a]->a; [b]->a, a->a,
or even a are correct types, but too general, whereas something like [Integer]->Integer is too
speci�c. The existence of unique principal types is the hallmark feature of the Hindley-Milner type
system, which forms the basis of the type systems of Haskell, ML, Miranda,4 and several other
(mostly functional) languages.

2.2 User-De�ned Types

We can de�ne our own types in Haskell using a data declaration, which we introduce via a series
of examples (x4.2.1).

An important prede�ned type in Haskell is that of truth values:

data Bool = False | True

The type being de�ned here is Bool, and it has exactly two values: True and False. Type Bool is
an example of a (nullary) type constructor, and True and False are (also nullary) data constructors
(or just constructors, for short).

Similarly, we might wish to de�ne a color type:

data Color = Red | Green | Blue | Indigo | Violet

Both Bool and Color are examples of enumerated types, since they consist of a �nite number of
nullary data constructors.

Here is an example of a type with just one data constructor:

data Point a = Pt a a

Because of the single constructor, a type like Point is often called a tuple type, since it is essentially

4\Miranda" is a trademark of Research Software, Ltd.

just a cartesian product (in this case binary) of other types.5 In contrast, multi-constructor types,
such as Bool and Color, are called (disjoint) union or sum types.

More importantly, however, Point is an example of a polymorphic type: for any type t, it
de�nes the type of cartesian points that use t as the coordinate type. The Point type can now be
seen clearly as a unary type constructor, since from the type t it constructs a new type Point t.
(In the same sense, using the list example given earlier, [] is also a type constructor. Given any
type t we can \apply" [] to yield a new type [t]. The Haskell syntax allows [] t to be written
as [t]. Similarly, -> is a type constructor: given two types t and u, t->u is the type of functions
mapping elements of type t to elements of type u.)

Note that the type of the binary data constructor Pt is a -> a -> Point a, and thus the
following typings are valid:

Pt 2.0 3.0 :: Point Float

Pt 'a' 'b' :: Point Char

Pt True False :: Point Bool

On the other hand, an expression such as Pt 'a' 1 is ill-typed because 'a' and 1 are of di�erent
types.

It is important to distinguish between applying a data constructor to yield a value, and applying
a type constructor to yield a type; the former happens at run-time and is how we compute things
in Haskell, whereas the latter happens at compile-time and is part of the type system's process of
ensuring type safety.

[Type constructors such as Point and data constructors such as Pt are in separate namespaces.
This allows the same name to be used for both a type constructor and data constructor, as in the
following:

data Point a = Point a a

While this may seem a little confusing at �rst, it serves to make the link between a type and its
data constructor more obvious.]

2.2.1 Recursive Types

Types can also be recursive, as in the type of binary trees:

data Tree a = Leaf a | Branch (Tree a) (Tree a)

Here we have de�ned a polymorphic binary tree type whose elements are either leaf nodes containing
a value of type a, or internal nodes (\branches") containing (recursively) two sub-trees.

When reading data declarations such as this, remember again that Tree is a type constructor,
whereas Branch and Leaf are data constructors. Aside from establishing a connection between
these constructors, the above declaration is essentially de�ning the following types for Branch and
Leaf:

Branch :: Tree a -> Tree a -> Tree a

Leaf :: a -> Tree a
5Tuples are somewhat like records in other languages.

With this example we have de�ned a type suÆciently rich to allow de�ning some interesting
(recursive) functions that use it. For example, suppose we wish to de�ne a function fringe that
returns a list of all the elements in the leaves of a tree from left to right. It's usually helpful to write
down the type of new functions �rst; in this case we see that the type should be Tree a -> [a].
That is, fringe is a polymorphic function that, for any type a, maps trees of a into lists of a. A
suitable de�nition follows:

fringe :: Tree a -> [a]

fringe (Leaf x) = [x]

fringe (Branch left right) = fringe left ++ fringe right

Here ++ is the in�x operator that concatenates two lists (its full de�nition will be given in Section
9.1). As with the length example given earlier, the fringe function is de�ned using pattern
matching, except that here we see patterns involving user-de�ned constructors: Leaf and Branch.
[Note that the formal parameters are easily identi�ed as the ones beginning with lower-case letters.]

2.3 Type Synonyms

For convenience, Haskell provides a way to de�ne type synonyms; i.e. names for commonly used
types. Type synonyms are created using a type declaration (x4.2.2). Here are several examples:

type String = [Char]

type Person = (Name,Address)

type Name = String

data Address = None | Addr String

Type synonyms do not de�ne new types, but simply give new names for existing types. For
example, the type Person -> Name is precisely equivalent to (String,Address) -> String. The
new names are often shorter than the types they are synonymous with, but this is not the only
purpose of type synonyms: they can also improve readability of programs by being more mnemonic;
indeed, the above examples highlight this. We can even give new names to polymorphic types:

type AssocList a b = [(a,b)]

This is the type of \association lists" which associate values of type a with those of type b.

2.4 Built-in Types Are Not Special

Earlier we introduced several \built-in" types such as lists, tuples, integers, and characters. We have
also shown how new user-de�ned types can be de�ned. Aside from special syntax, are the built-in
types in any way more special than the user-de�ned ones? The answer is no. The special syntax is
for convenience and for consistency with historical convention, but has no semantic consequences.

We can emphasize this point by considering what the type declarations would look like for these
built-in types if in fact we were allowed to use the special syntax in de�ning them. For example,
the Char type might be written as:

data Char = 'a' | 'b' | 'c' | ... -- This is not valid

| 'A' | 'B' | 'C' | ... -- Haskell code!

| '1' | '2' | '3' | ...

...

These constructor names are not syntactically valid; to �x them we would have to write something
like:

data Char = Ca | Cb | Cc | ...

| CA | CB | CC | ...

| C1 | C2 | C3 | ...

...

Even though these constructors are more concise, they are quite unconventional for representing
characters.

In any case, writing \pseudo-Haskell" code in this way helps us to see through the special
syntax. We see now that Char is just an enumerated type consisting of a large number of nullary
constructors. Thinking of Char in this way makes it clear that we can pattern-match against
characters in function de�nitions, just as we would expect to be able to do so for any of a type's
constructors.

[This example also demonstrates the use of comments in Haskell; the characters -- and all
subsequent characters to the end of the line are ignored. Haskell also permits nested comments
which have the form {-: : :-} and can appear anywhere (x2.2).]

Similarly, we could de�ne Int (�xed precision integers) and Integer by:

data Int = -65532 | ... | -1 | 0 | 1 | ... | 65532 -- more pseudo-code

data Integer = ... -2 | -1 | 0 | 1 | 2 ...

where -65532 and 65532, say, are the maximum and minimum �xed precision integers for a given
implementation. Int is a much larger enumeration than Char, but it's still �nite! In contrast, the
pseudo-code for Integer is intended to convey an in�nite enumeration.

Tuples are also easy to de�ne playing this game:

data (a,b) = (a,b) -- more pseudo-code

data (a,b,c) = (a,b,c)

data (a,b,c,d) = (a,b,c,d)

. .

. .

. .

Each declaration above de�nes a tuple type of a particular length, with (...) playing a role in
both the expression syntax (as data constructor) and type-expression syntax (as type constructor).
The vertical dots after the last declaration are intended to convey an in�nite number of such
declarations, re
ecting the fact that tuples of all lengths are allowed in Haskell.

Lists are also easily handled, and more interestingly, they are recursive:

data [a] = [] | a : [a] -- more pseudo-code

We can now see clearly what we described about lists earlier: [] is the empty list, and : is the in�x

list constructor; thus [1,2,3] must be equivalent to the list 1:2:3:[]. (: is right associative.)
The type of [] is [a], and the type of : is a->[a]->[a].

[The way \:" is de�ned here is actually legal syntax|in�x constructors are permitted in data

declarations, and are distinguished from in�x operators (for pattern-matching purposes) by the fact
that they must begin with a \:" (a property trivially satis�ed by \:").]

At this point the reader should note carefully the di�erences between tuples and lists, which
the above de�nitions make abundantly clear. In particular, note the recursive nature of the list
type whose elements are homogeneous and of arbitrary length, and the non-recursive nature of a
(particular) tuple type whose elements are heterogeneous and of �xed length. The typing rules for
tuples and lists should now also be clear:

For (e1,e2, : : : ,en); n � 2, if ti is the type of ei, then the type of the tuple is (t1,t2, : : : ,tn).

For [e1,e2, : : : ,en]; n � 0, each ei must have the same type t, and the type of the list is [t].

2.4.1 List Comprehensions and Arithmetic Sequences

As with Lisp dialects, lists are pervasive in Haskell, and as with other functional languages, there is
yet more syntactic sugar to aid in their creation. Aside from the constructors for lists just discussed,
Haskell provides an expression known as a list comprehension that is best explained by example:

[ f x | x <- xs ]

This expression can intuitively be read as \the list of all f x such that x is drawn from xs." The
similarity to set notation is not a coincidence. The phrase x <- xs is called a generator, of which
more than one is allowed, as in:

[ (x,y) | x <- xs, y <- ys ]

This list comprehension forms the cartesian product of the two lists xs and ys. The elements are
selected as if the generators were \nested" from left to right (with the rightmost generator varying
fastest); thus, if xs is [1,2] and ys is [3,4], the result is [(1,3),(1,4),(2,3),(2,4)].

Besides generators, boolean expressions called guards are permitted. Guards place constraints
on the elements generated. For example, here is a concise de�nition of everybody's favorite sorting
algorithm:

quicksort [] = []

quicksort (x:xs) = quicksort [y | y <- xs, y<x ]

++ [x]

++ quicksort [y | y <- xs, y>=x]

To further support the use of lists, Haskell has special syntax for arithmetic sequences, which
are best explained by a series of examples:

[1..10] ) [1,2,3,4,5,6,7,8,9,10]

[1,3..10] ) [1,3,5,7,9]

[1,3..] ) [1,3,5,7,9, ... (in�nite sequence)

More will be said about arithmetic sequences in Section 8.2, and \in�nite lists" in Section 3.4.

2.4.2 Strings

As another example of syntactic sugar for built-in types, we note that the literal string "hello" is
actually shorthand for the list of characters ['h','e','l','l','o']. Indeed, the type of "hello"
is String, where String is a prede�ned type synonym (that we gave as an earlier example):

type String = [Char]

This means we can use prede�ned polymorphic list functions to operate on strings. For example:

"hello" ++ " world" ) "hello world"

3 Functions

Since Haskell is a functional language, one would expect functions to play a major role, and indeed
they do. In this section, we look at several aspects of functions in Haskell.

First, consider this de�nition of a function which adds its two arguments:

add :: Integer -> Integer -> Integer

add x y = x + y

This is an example of a curried function.6 An application of add has the form add e1 e2, and
is equivalent to (add e1) e2, since function application associates to the left. In other words,
applying add to one argument yields a new function which is then applied to the second argu-
ment. This is consistent with the type of add, Integer->Integer->Integer, which is equivalent
to Integer->(Integer->Integer); i.e. -> associates to the right. Indeed, using add, we can de�ne
inc in a di�erent way from earlier:

inc = add 1

This is an example of the partial application of a curried function, and is one way that a function
can be returned as a value. Let's consider a case in which it's useful to pass a function as an
argument. The well-known map function is a perfect example:

map :: (a->b) -> [a] -> [b]

map f [] = []

map f (x:xs) = f x : map f xs

[Function application has higher precedence than any in�x operator, and thus the right-hand side
of the second equation parses as (f x) : (map f xs).] The map function is polymorphic and
its type indicates clearly that its �rst argument is a function; note also that the two a's must be
instantiated with the same type (likewise for the b's). As an example of the use of map, we can
increment the elements in a list:

map (add 1) [1,2,3] ) [2,3,4]

6The name curry derives from the person who popularized the idea: Haskell Curry. To get the e�ect of an
uncurried function, we could use a tuple, as in:

add (x,y) = x + y

But then we see that this version of add is really just a function of one argument!

These examples demonstrate the �rst-class nature of functions, which when used in this way
are usually called higher-order functions.

3.1 Lambda Abstractions

Instead of using equations to de�ne functions, we can also de�ne them \anonymously" via a lambda
abstraction. For example, a function equivalent to inc could be written as \x -> x+1. Similarly,
the function add is equivalent to \x -> \y -> x+y. Nested lambda abstractions such as this may
be written using the equivalent shorthand notation \x y -> x+y. In fact, the equations:

inc x = x+1

add x y = x+y

are really shorthand for:

inc = \x -> x+1

add = \x y -> x+y

We will have more to say about such equivalences later.

In general, given that x has type t1 and exp has type t2, then \x->exp has type t1->t2.

3.2 In�x Operators

In�x operators are really just functions, and can also be de�ned using equations. For example, here
is a de�nition of a list concatenation operator:

(++) :: [a] -> [a] -> [a]

[] ++ ys = ys

(x:xs) ++ ys = x : (xs++ys)

[Lexically, in�x operators consist entirely of \symbols," as opposed to normal identi�ers which are
alphanumeric (x2.4). Haskell has no pre�x operators, with the exception of minus (-), which is
both in�x and pre�x.]

As another example, an important in�x operator on functions is that for function composition:

(.) :: (b->c) -> (a->b) -> (a->c)

f . g = \ x -> f (g x)

3.2.1 Sections

Since in�x operators are really just functions, it makes sense to be able to partially apply them as
well. In Haskell the partial application of an in�x operator is called a section. For example:

(x+) � \y -> x+y

(+y) � \x -> x+y

(+) � \x y -> x+y

[The parentheses are mandatory.]

The last form of section given above essentially coerces an in�x operator into an equivalent
functional value, and is handy when passing an in�x operator as an argument to a function, as
in map (+) [1,2,3] (the reader should verify that this returns a list of functions!). It is also
necessary when giving a function type signature, as in the examples of (++) and (.) given earlier.

We can now see that add de�ned earlier is just (+), and inc is just (+1)! Indeed, these
de�nitions would do just �ne:

inc = (+ 1)

add = (+)

We can coerce an in�x operator into a functional value, but can we go the other way? Yes|we
simply enclose an identi�er bound to a functional value in backquotes. For example, x `add` y

is the same as add x y.7 Some functions read better this way. An example is the prede�ned list
membership predicate elem; the expression x `elem` xs can be read intuitively as \x is an element
of xs."

[There are some special rules regarding sections involving the pre�x/in�x operator -; see
(x3.5,x3.4).]

At this point, the reader may be confused at having so many ways to de�ne a function! The
decision to provide these mechanisms partly re
ects historical conventions, and partly re
ects the
desire for consistency (for example, in the treatment of in�x vs. regular functions).

3.2.2 Fixity Declarations

A �xity declaration can be given for any in�x operator or constructor (including those made from
ordinary identi�ers, such as `elem`). This declaration speci�es a precedence level from 0 to 9 (with
9 being the strongest; normal application is assumed to have a precedence level of 10), and left-,
right-, or non-associativity. For example, the �xity declarations for ++ and . are:

infixr 5 ++

infixr 9 .

Both of these specify right-associativity, the �rst with a precedence level of 5, the other 9. Left
associativity is speci�ed via infixl, and non-associativity by infix. Also, the �xity of more than
one operator may be speci�ed with the same �xity declaration. If no �xity declaration is given for
a particular operator, it defaults to infixl 9. (See x5.9 for a detailed de�nition of the associativity
rules.)

3.3 Functions are Non-strict

Suppose bot is de�ned by:

7Note carefully that add is enclosed in backquotes, not apostrophes as used in the syntax of characters; i.e. 'f' is
a character, whereas `f` is an in�x operator. Fortunately, most ASCII terminals distinguish these much better than
the font used in this manuscript.

bot = bot

In other words, bot is a non-terminating expression. Abstractly, we denote the value of a non-
terminating expression as ? (read \bottom"). Expressions that result in some kind of a run-time
error, such as 1/0, also have this value. Such an error is not recoverable: programs will not continue
past these errors. Errors encountered by the I/O system, such as an end-of-�le error, are recoverable
and are handled in a di�erent manner. (Such an I/O error is really not an error at all but rather
an exception. Much more will be said about exceptions in Section 7.)

A function f is said to be strict if, when applied to a nonterminating expression, it also fails to
terminate. In other words, f is strict i� the value of f bot is ?. For most programming languages,
all functions are strict. But this is not so in Haskell. As a simple example, consider const1, the
constant 1 function, de�ned by:

const1 x = 1

The value of const1 bot in Haskell is 1. Operationally speaking, since const1 does not \need"
the value of its argument, it never attempts to evaluate it, and thus never gets caught in a nonter-
minating computation. For this reason, non-strict functions are also called \lazy functions", and
are said to evaluate their arguments \lazily", or \by need".

Since error and nonterminating values are semantically the same in Haskell, the above argument
also holds for errors. For example, const1 (1/0) also evaluates properly to 1.

Non-strict functions are extremely useful in a variety of contexts. The main advantage is that
they free the programmer from many concerns about evaluation order. Computationally expensive
values may be passed as arguments to functions without fear of them being computed if they are
not needed. An important example of this is a possibly in�nite data structure.

Another way of explaining non-strict functions is that Haskell computes using de�nitions rather
than the assignments found in traditional languages. Read a declaration such as

v = 1/0

as `de�ne v as 1/0' instead of `compute 1/0 and store the result in v'. Only if the value (de�nition)
of v is needed will the division by zero error occur. By itself, this declaration does not imply
any computation. Programming using assignments requires careful attention to the ordering of
the assignments: the meaning of the program depends on the order in which the assignments are
executed. De�nitions, in contrast, are much simpler: they can be presented in any order without
a�ecting the meaning of the program.

3.4 \In�nite" Data Structures

One advantage of the non-strict nature of Haskell is that data constructors are non-strict, too. This
should not be surprising, since constructors are really just a special kind of function (the distin-
guishing feature being that they can be used in pattern matching). For example, the constructor
for lists, (:), is non-strict.

Non-strict constructors permit the de�nition of (conceptually) in�nite data structures. Here is
an in�nite list of ones:

Figure 1: Circular Fibonacci Sequence

ones = 1 : ones

Perhaps more interesting is the function numsFrom:

numsFrom n = n : numsFrom (n+1)

Thus numsFrom n is the in�nite list of successive integers beginning with n. From it we can construct
an in�nite list of squares:

squares = map (^2) (numsfrom 0)

(Note the use of a section; ^ is the in�x exponentiation operator.)

Of course, eventually we expect to extract some �nite portion of the list for actual computation,
and there are lots of prede�ned functions in Haskell that do this sort of thing: take, takeWhile,
filter, and others. The de�nition of Haskell includes a large set of built-in functions and types|
this is called the \Standard Prelude". The complete Standard Prelude is included in Appendix A
of the Haskell report; see the portion named PreludeList for many useful functions involving lists.
For example, take removes the �rst n elements from a list:

take 5 squares ) [0,1,4,9,16]

The de�nition of ones above is an example of a circular list. In most circumstances laziness
has an important impact on eÆciency, since an implementation can be expected to implement the
list as a true circular structure, thus saving space.

For another example of the use of circularity, the Fibonacci sequence can be computed eÆciently
as the following in�nite sequence:

fib = 1 : 1 : [ a+b | (a,b) <- zip fib (tail fib) ]

where zip is a Standard Prelude function that returns the pairwise interleaving of its two list
arguments:

zip (x:xs) (y:ys) = (x,y) : zip xs ys

zip xs ys = []

Note how fib, an in�nite list, is de�ned in terms of itself, as if it were \chasing its tail." Indeed,
we can draw a picture of this computation as shown in Figure 1.

For another application of in�nite lists, see Section 4.4.

3.5 The Error Function

Haskell has a built-in function called error whose type is String->a. This is a somewhat odd
function: From its type it looks as if it is returning a value of a polymorphic type about which it
knows nothing, since it never receives a value of that type as an argument!

In fact, there is one value \shared" by all types: ?. Indeed, semantically that is exactly what
value is always returned by error (recall that all errors have value ?). However, we can expect that
a reasonable implementation will print the string argument to error for diagnostic purposes. Thus
this function is useful when we wish to terminate a program when something has \gone wrong."
For example, the actual de�nition of head taken from the Standard Prelude is:

head (x:xs) = x

head [] = error "head{PreludeList}: head []"

4 Case Expressions and Pattern Matching

Earlier we gave several examples of pattern matching in de�ning functions|for example length

and fringe. In this section we will look at the pattern-matching process in greater detail (x3.17).8

Patterns are not \�rst-class;" there is only a �xed set of di�erent kinds of patterns. We have
already seen several examples of data constructor patterns; both length and fringe de�ned earlier
use such patterns, the former on the constructors of a \built-in" type (lists), the latter on a user-
de�ned type (Tree). Indeed, matching is permitted using the constructors of any type, user-de�ned
or not. This includes tuples, strings, numbers, characters, etc. For example, here's a contrived
function that matches against a tuple of \constants:"

contrived :: ([a], Char, (Int, Float), String, Bool) -> Bool

contrived ([], 'b', (1, 2.0), "hi", True) = False

This example also demonstrates that nesting of patterns is permitted (to arbitrary depth).

Technically speaking, formal parameters9 are also patterns|it's just that they never fail to
match a value. As a \side e�ect" of the successful match, the formal parameter is bound to the
value it is being matched against. For this reason patterns in any one equation are not allowed
to have more than one occurrence of the same formal parameter (a property called linearity x3.17,
x3.3, x4.4.2).

Patterns such as formal parameters that never fail to match are said to be irrefutable, in contrast
to refutable patterns which may fail to match. The pattern used in the contrived example above
is refutable. There are three other kinds of irrefutable patterns, two of which we will introduce now
(the other we will delay until Section 4.4).

8Pattern matching in Haskell is di�erent from that found in logic programming languages such as Prolog; in
particular, it can be viewed as \one-way" matching, whereas Prolog allows \two-way" matching (via uni�cation),
along with implicit backtracking in its evaluation mechanism.

9The Report calls these variables.

As-patterns. Sometimes it is convenient to name a pattern for use on the right-hand side of an
equation. For example, a function that duplicates the �rst element in a list might be written as:

f (x:xs) = x:x:xs

(Recall that \:" associates to the right.) Note that x:xs appears both as a pattern on the left-hand
side, and an expression on the right-hand side. To improve readability, we might prefer to write
x:xs just once, which we can achieve using an as-pattern as follows:10

f s@(x:xs) = x:s

Technically speaking, as-patterns always result in a successful match, although the sub-pattern (in
this case x:xs) could, of course, fail.

Wild-cards. Another common situation is matching against a value we really care nothing about.
For example, the functions head and tail de�ned in Section 2.1 can be rewritten as:

head (x:_) = x

tail (_:xs) = xs

in which we have \advertised" the fact that we don't care what a certain part of the input is.
Each wild-card independently matches anything, but in contrast to a formal parameter, each binds
nothing; for this reason more than one is allowed in an equation.

4.1 Pattern-Matching Semantics

So far we have discussed how individual patterns are matched, how some are refutable, some are
irrefutable, etc. But what drives the overall process? In what order are the matches attempted?
What if none succeeds? This section addresses these questions.

Pattern matching can either fail, succeed or diverge. A successful match binds the formal
parameters in the pattern. Divergence occurs when a value needed by the pattern contains an error
(?). The matching process itself occurs \top-down, left-to-right." Failure of a pattern anywhere
in one equation results in failure of the whole equation, and the next equation is then tried. If all
equations fail, the value of the function application is ?, and results in a run-time error.

For example, if [1,2] is matched against [0,bot], then 1 fails to match 0, so the result is a
failed match. (Recall that bot, de�ned earlier, is a variable bound to ?.) But if [1,2] is matched
against [bot,0], then matching 1 against bot causes divergence (i.e. ?).

The other twist to this set of rules is that top-level patterns may also have a boolean guard, as
in this de�nition of a function that forms an abstract version of a number's sign:

sign x | x > 0 = 1

| x == 0 = 0

| x < 0 = -1

Note that a sequence of guards may be provided for the same pattern; as with patterns, they are
evaluated top-down, and the �rst that evaluates to True results in a successful match.

10Another advantage to doing this is that a naive implementation might completely reconstruct x:xs rather than
re-use the value being matched against.

4.2 An Example

The pattern-matching rules can have subtle e�ects on the meaning of functions. For example,
consider this de�nition of take:

take 0 _ = []

take _ [] = []

take n (x:xs) = x : take (n-1) xs

and this slightly di�erent version (the �rst 2 equations have been reversed):

take1 _ [] = []

take1 0 _ = []

take1 n (x:xs) = x : take1 (n-1) xs

Now note the following:

take 0 bot ) []

take1 0 bot ) ?

take bot [] ) ?

take1 bot [] ) []

We see that take is \more de�ned" with respect to its second argument, whereas take1 is more
de�ned with respect to its �rst. It is diÆcult to say in this case which de�nition is better. Just
remember that in certain applications, it may make a di�erence. (The Standard Prelude includes
a de�nition corresponding to take.)

4.3 Case Expressions

Pattern matching provides a way to \dispatch control" based on structural properties of a value.
In many circumstances we don't wish to de�ne a function every time we need to do this, but
so far we have only shown how to do pattern matching in function de�nitions. Haskell's case
expression provides a way to solve this problem. Indeed, the meaning of pattern matching in
function de�nitions is speci�ed in the Report in terms of case expressions, which are considered
more primitive. In particular, a function de�nition of the form:

f p11 : : : p1k = e1
: : :

f pn1 : : : pnk = en

where each pij is a pattern, is semantically equivalent to:

f x1 x2 : : : xk = case (x1, : : : , xk) of (p11 ; : : : ; p1k) -> e1
: : :

(pn1 ; : : : ; pnk) -> en

where the xi are new identi�ers. (For a more general translation that includes guards, see x4.4.2.)
For example, the de�nition of take given earlier is equivalent to:

take m ys = case (m,ys) of

(0,_) -> []

(_,[]) -> []

(n,x:xs) -> x : take (n-1) xs

A point not made earlier is that, for type correctness, the types of the right-hand sides of a case
expression or set of equations comprising a function de�nition must all be the same; more precisely,
they must all share a common principal type.

The pattern-matching rules for case expressions are the same as we have given for function
de�nitions, so there is really nothing new to learn here, other than to note the convenience that
case expressions o�er. Indeed, there's one use of a case expression that is so common that it has
special syntax: the conditional expression. In Haskell, conditional expressions have the familiar
form:

if e1 then e2 else e3

which is really short-hand for:
case e1 of True -> e2

False -> e3

From this expansion it should be clear that e1 must have type Bool, and e2 and e3 must have the
same (but otherwise arbitrary) type. In other words, if-then-else when viewed as a function has
type Bool->a->a->a.

4.4 Lazy Patterns

There is one other kind of pattern allowed in Haskell. It is called a lazy pattern, and has the form
~pat. Lazy patterns are irrefutable: matching a value v against ~pat always succeeds, regardless
of pat. Operationally speaking, if an identi�er in pat is later \used" on the right-hand-side, it will
be bound to that portion of the value that would result if v were to successfully match pat, and ?
otherwise.

Lazy patterns are useful in contexts where in�nite data structures are being de�ned recursively.
For example, in�nite lists are an excellent vehicle for writing simulation programs, and in this
context the in�nite lists are often called streams. Consider the simple case of simulating the
interactions between a server process server and a client process client, where client sends a
sequence of requests to server, and server replies to each request with some kind of response.
This situation is shown pictorially in Figure 2. (Note that client also takes an initial message as
argument.) Using streams to simulate the message sequences, the Haskell code corresponding to
this diagram is:

reqs = client init resps

resps = server reqs

These recursive equations are a direct lexical transliteration of the diagram.

Let us further assume that the structure of the server and client look something like this:

client init (resp:resps) = init : client (next resp) resps

server (req:reqs) = process req : server reqs

Figure 2: Client-Server Simulation

where we assume that next is a function that, given a response from the server, determines the
next request, and process is a function that processes a request from the client, returning an
appropriate response.

Unfortunately, this program has a serious problem: it will not produce any output! The problem
is that client, as used in the recursive setting of reqs and resps, attempts a match on the response
list before it has submitted its �rst request! In other words, the pattern matching is being done
\too early." One way to �x this is to rede�ne client as follows:

client init resps = init : client (next (head resps)) (tail resps)

Although workable, this solution does not read as well as that given earlier. A better solution is to
use a lazy pattern:

client init ~(resp:resps) = init : client (next resp) resps

Because lazy patterns are irrefutable, the match will immediately succeed, allowing the initial
request to be \submitted", in turn allowing the �rst response to be generated; the engine is now
\primed", and the recursion takes care of the rest.

As an example of this program in action, if we de�ne:

init = 0

next resp = resp

process req = req+1

then we see that:

take 10 reqs ) [0,1,2,3,4,5,6,7,8,9]

As another example of the use of lazy patterns, consider the de�nition of Fibonacci given earlier:

fib = 1 : 1 : [ a+b | (a,b) <- zip fib (tail fib) ]

We might try rewriting this using an as-pattern:

fib@(1:tfib) = 1 : 1 : [ a+b | (a,b) <- zip fib tfib ]

This version of fib has the (small) advantage of not using tail on the right-hand side, since it is
available in \destructured" form on the left-hand side as tfib.

[This kind of equation is called a pattern binding because it is a top-level equation in which the
entire left-hand side is a pattern; i.e. both fib and tfib become bound within the scope of the
declaration.]

Now, using the same reasoning as earlier, we should be led to believe that this program will
not generate any output. Curiously, however, it does, and the reason is simple: in Haskell, pattern
bindings are assumed to have an implicit ~ in front of them, re
ecting the most common behavior
expected of pattern bindings, and avoiding some anomalous situations which are beyond the scope
of this tutorial. Thus we see that lazy patterns play an important role in Haskell, if only implicitly.

4.5 Lexical Scoping and Nested Forms

It is often desirable to create a nested scope within an expression, for the purpose of creating local
bindings not seen elsewhere|i.e. some kind of \block-structuring" form. In Haskell there are two
ways to achieve this:

Let Expressions. Haskell's let expressions are useful whenever a nested set of bindings is re-
quired. As a simple example, consider:

let y = a*b

f x = (x+y)/y

in f c + f d

The set of bindings created by a let expression is mutually recursive, and pattern bindings are
treated as lazy patterns (i.e. they carry an implicit ~). The only kind of declarations permitted
are type signatures, function bindings, and pattern bindings.

Where Clauses. Sometimes it is convenient to scope bindings over several guarded equations,
which requires a where clause:

f x y | y>z = ...

| y==z = ...

| y<z = ...

where z = x*x

Note that this cannot be done with a let expression, which only scopes over the expression which
it encloses. A where clause is only allowed at the top level of a set of equations or case expression.
The same properties and constraints on bindings in let expressions apply to those in where clauses.

These two forms of nested scope seem very similar, but remember that a let expression is an
expression, whereas a where clause is not|it is part of the syntax of function declarations and case
expressions.

4.6 Layout

The reader may have been wondering how it is that Haskell programs avoid the use of semicolons,
or some other kind of terminator, to mark the end of equations, declarations, etc. For example,
consider this let expression from the last section:

let y = a*b

f x = (x+y)/y

in f c + f d

How does the parser know not to parse this as:

let y = a*b f

x = (x+y)/y

in f c + f d

?

The answer is that Haskell uses a two-dimensional syntax called layout that essentially relies
on declarations being \lined up in columns." In the above example, note that y and f begin in
the same column. The rules for layout are spelled out in detail in the Report (x2.7, xB.3), but in
practice, use of layout is rather intuitive. Just remember two things:

First, the next character following any of the keywords where, let, or of is what determines
the starting column for the declarations in the where, let, or case expression being written (the
rule also applies to where used in the class and instance declarations to be introduced in Section
5). Thus we can begin the declarations on the same line as the keyword, the next line, etc. (The
do keyword, to be discussed later, also uses layout).

Second, just be sure that the starting column is further to the right than the starting column
associated with the immediately surrounding clause (otherwise it would be ambiguous). The \ter-
mination" of a declaration happens when something appears at or to the left of the starting column
associated with that binding form.11

Layout is actually shorthand for an explicit grouping mechanism, which deserves mention be-
cause it can be useful under certain circumstances. The let example above is equivalent to:

let { y = a*b

; f x = (x+y)/y

}

in f c + f d

Note the explicit curly braces and semicolons. One way in which this explicit notation is useful is
when more than one declaration is desired on a line; for example, this is a valid expression:

let y = a*b; z = a/b

f x = (x+y)/z

in f c + f d

For another example of the expansion of layout into explicit delimiters, see x2.7.

The use of layout greatly reduces the syntactic clutter associated with declaration lists, thus
enhancing readability. It is easy to learn, and its use is encouraged.

5 Type Classes and Overloading

There is one �nal feature of Haskell's type system that sets it apart from other programming lan-
guages. The kind of polymorphism that we have talked about so far is commonly called parametric
polymorphism. There is another kind called ad hoc polymorphism, better known as overloading.
Here are some examples of ad hoc polymorphism:

11Haskell observes the convention that tabs count as 8 blanks; thus care must be taken when using an editor which
may observe some other convention.

� The literals 1, 2, etc. are often used to represent both �xed and arbitrary precision integers.

� Numeric operators such as + are often de�ned to work on many di�erent kinds of numbers.

� The equality operator (== in Haskell) usually works on numbers and many other (but not all)
types.

Note that these overloaded behaviors are di�erent for each type (in fact the behavior is sometimes
unde�ned, or error), whereas in parametric polymorphism the type truly does not matter (fringe,
for example, really doesn't care what kind of elements are found in the leaves of a tree). In Haskell,
type classes provide a structured way to control ad hoc polymorphism, or overloading.

Let's start with a simple, but important, example: equality. There are many types for which we
would like equality de�ned, but some for which we would not. For example, comparing the equality
of functions is generally considered computationally intractable, whereas we often want to compare
two lists for equality.12 To highlight the issue, consider this de�nition of the function elem which
tests for membership in a list:

x `elem` [] = False

x `elem` (y:ys) = x==y || (x `elem` ys)

[For the stylistic reason we discussed in Section 3.1, we have chosen to de�ne elem in in�x form.
== and || are the in�x operators for equality and logical or, respectively.]

Intuitively speaking, the type of elem \ought" to be: a->[a]->Bool. But this would imply that ==
has type a->a->Bool, even though we just said that we don't expect == to be de�ned for all types.

Furthermore, as we have noted earlier, even if == were de�ned on all types, comparing two
lists for equality is very di�erent from comparing two integers. In this sense, we expect == to be
overloaded to carry on these various tasks.

Type classes conveniently solve both of these problems. They allow us to declare which types
are instances of which class, and to provide de�nitions of the overloaded operations associated with
a class. For example, let's de�ne a type class containing an equality operator:

class Eq a where

(==) :: a -> a -> Bool

Here Eq is the name of the class being de�ned, and == is the single operation in the class. This
declaration may be read \a type a is an instance of the class Eq if there is an (overloaded) operation
==, of the appropriate type, de�ned on it." (Note that == is only de�ned on pairs of objects of the
same type.)

The constraint that a type a must be an instance of the class Eq is written Eq a. Thus Eq a

is not a type expression, but rather it expresses a constraint on a type, and is called a context.
Contexts are placed at the front of type expressions. For example, the e�ect of the above class
declaration is to assign the following type to ==:

(==) :: (Eq a) => a -> a -> Bool

12The kind of equality we are referring to here is \value equality," and opposed to the \pointer equality" found,
for example, with Java's ==. Pointer equality is not referentially transparent, and thus does not sit well in a purely
functional language.

This should be read, \For every type a that is an instance of the class Eq, == has type a->a->Bool".
This is the type that would be used for == in the elem example, and indeed the constraint imposed
by the context propagates to the principal type for elem:

elem :: (Eq a) => a -> [a] -> Bool

This is read, \For every type a that is an instance of the class Eq, elem has type a->[a]->Bool".
This is just what we want|it expresses the fact that elem is not de�ned on all types, just those
for which we know how to compare elements for equality.

So far so good. But how do we specify which types are instances of the class Eq, and the actual
behavior of == on each of those types? This is done with an instance declaration. For example:

instance Eq Integer where

x == y = x `integerEq` y

The de�nition of == is called a method. The function integerEq happens to be the primitive
function that compares integers for equality, but in general any valid expression is allowed on the
right-hand side, just as for any other function de�nition. The overall declaration is essentially
saying: \The type Integer is an instance of the class Eq, and here is the de�nition of the method
corresponding to the operation ==." Given this declaration, we can now compare �xed precision
integers for equality using ==. Similarly:

instance Eq Float where

x == y = x `floatEq` y

allows us to compare 
oating point numbers using ==.

Recursive types such as Tree de�ned earlier can also be handled:

instance (Eq a) => Eq (Tree a) where

Leaf a == Leaf b = a == b

(Branch l1 r1) == (Branch l2 r2) = (l1==l2) && (r1==r2)

_ == _ = False

Note the context Eq a in the �rst line|this is necessary because the elements in the leaves (of type
a) are compared for equality in the second line. The additional constraint is essentially saying that
we can compare trees of a's for equality as long as we know how to compare a's for equality. If the
context were omitted from the instance declaration, a static type error would result.

The Haskell Report, especially the Prelude, contains a wealth of useful examples of type classes.
Indeed, a class Eq is de�ned that is slightly larger than the one de�ned earlier:

class Eq a where

(==), (/=) :: a -> a -> Bool

x /= y = not (x == y)

This is an example of a class with two operations, one for equality, the other for inequality. It also
demonstrates the use of a default method, in this case for the inequality operation /=. If a method
for a particular operation is omitted in an instance declaration, then the default one de�ned in
the class declaration, if it exists, is used instead. For example, the three instances of Eq de�ned
earlier will work perfectly well with the above class declaration, yielding just the right de�nition of
inequality that we want: the logical negation of equality.

Haskell also supports a notion of class extension. For example, we may wish to de�ne a class
Ord which inherits all of the operations in Eq, but in addition has a set of comparison operations
and minimum and maximum functions:

class (Eq a) => Ord a where

(<), (<=), (>=), (>) :: a -> a -> Bool

max, min :: a -> a -> a

Note the context in the class declaration. We say that Eq is a superclass of Ord (conversely, Ord
is a subclass of Eq), and any type which is an instance of Ord must also be an instance of Eq. (In
the next Section we give a fuller de�nition of Ord taken from the Prelude.)

One bene�t of such class inclusions is shorter contexts: a type expression for a function that
uses operations from both the Eq and Ord classes can use the context (Ord a), rather than
(Eq a, Ord a), since Ord \implies" Eq. More importantly, methods for subclass operations can
assume the existence of methods for superclass operations. For example, the Ord declaration in the
Standard Prelude contains this default method for (<):

x < y = x <= y && x /= y

As an example of the use of Ord, the principal typing of quicksort de�ned in Section 2.4.1 is:

quicksort :: (Ord a) => [a] -> [a]

In other words, quicksort only operates on lists of values of ordered types. This typing for
quicksort arises because of the use of the comparison operators < and >= in its de�nition.

Haskell also permits multiple inheritance, since classes may have more than one superclass. For
example, the declaration

class (Eq a, Show a) => C a where ...

creates a class C which inherits operations from both Eq and Show.

Class methods are treated as top level declarations in Haskell. They share the same namespace
as ordinary variables; a name cannot be used to denote both a class method and a variable or
methods in di�erent classes.

Contexts are also allowed in data declarations; see x4.2.1.

Class methods may have additional class constraints on any type variable except the one de�ning
the current class. For example, in this class:

class C a where

m :: Show b => a -> b

the method m requires that type b is in class Show. However, the method m could not place any
additional class constraints on type a. These would instead have to be part of the context in the
class declaration.

So far, we have been using \�rst-order" types. For example, the type constructor Tree has so
far always been paired with an argument, as in Tree Integer (a tree containing Integer values)
or Tree a (representing the family of trees containing a values). But Tree by itself is a type
constructor, and as such takes a type as an argument and returns a type as a result. There are

no values in Haskell that have this type, but such \higher-order" types can be used in class

declarations.

To begin, consider the following Functor class (taken from the Prelude):

class Functor f where

fmap :: (a -> b) -> f a -> f b

The fmap function generalizes the map function used previously. Note that the type variable f is
applied to other types in f a and f b. Thus we would expect it to be bound to a type such as
Tree which can be applied to an argument. An instance of Functor for type Tree would be:

instance Functor Tree where

fmap f (Leaf x) = Leaf (f x)

fmap f (Branch t1 t2) = Branch (fmap f t1) (fmap f t2)

This instance declaration declares that Tree, rather than Tree a, is an instance of Functor. This
capability is quite useful, and here demonstrates the ability to describe generic \container" types,
allowing functions such as fmap to work uniformly over arbitrary trees, lists, and other data types.

[Type applications are written in the same manner as function applications. The type T a b is
parsed as (T a) b. Types such as tuples which use special syntax can be written in an alternative
style which allows currying. For functions, (->) is a type constructor; the types f -> g and
(->) f g are the same. Similarly, the types [a] and [] a are the same. For tuples, the type
constructors (as well as the data constructors) are (,), (,,), and so on.]

As we know, the type system detects typing errors in expressions. But what about errors due to
malformed type expressions? The expression (+) 1 2 3 results in a type error since (+) takes only
two arguments. Similarly, the type Tree Int Int should produce some sort of an error since the
Tree type takes only a single argument. So, how does Haskell detect malformed type expressions?
The answer is a second type system which ensures the correctness of types! Each type has an
associated kind which ensures that the type is used correctly.

Type expressions are classi�ed into di�erent kinds which take one of two possible forms:

� The symbol � represents the kind of type associated with concrete data objects. That is, if
the value v has type t , the kind of v must be �.

� If �1 and �2 are kinds, then �1 ! �2 is the kind of types that take a type of kind �1 and
return a type of kind �2.

The type constructor Tree has the kind � ! �; the type Tree Int has the kind �. Members of the
Functor class must all have the kind � ! �; a kinding error would result from an declaration such
as

instance Functor Integer where ...

since Integer has the kind �.

Kinds do not appear directly in Haskell programs. The compiler infers kinds before doing
type checking without any need for `kind declarations'. Kinds stay in the background of a Haskell
program except when an erroneous type signature leads to a kind error. Kinds are simple enough
that compilers should be able to provide descriptive error messages when kind con
icts occur. See
x4.1.1 and x4.6 for more information about kinds.

A Di�erent Perspective. Before going on to further examples of the use of type classes, it is
worth pointing out two other views of Haskell's type classes. The �rst is by analogy with object-
oriented programming (OOP). In the following general statement about OOP, simply substituting
type class for class, and type for object, yields a valid summary of Haskell's type class mechanism:

\Classes capture common sets of operations. A particular object may be an instance of a class,
and will have a method corresponding to each operation. Classes may be arranged hierarchically,
forming notions of superclasses and subclasses, and permitting inheritance of operations/methods.
A default method may also be associated with an operation."

In contrast to OOP, it should be clear that types are not objects, and in particular there is no
notion of an object's or type's internal mutable state. An advantage over some OOP languages is
that methods in Haskell are completely type-safe: any attempt to apply a method to a value whose
type is not in the required class will be detected at compile time instead of at runtime. In other
words, methods are not \looked up" at runtime but are simply passed as higher-order functions.

A di�erent perspective can be gotten by considering the relationship between parametric and ad
hoc polymorphism. We have shown how parametric polymorphism is useful in de�ning families of
types by universally quantifying over all types. Sometimes, however, that universal quanti�cation is
too broad|we wish to quantify over some smaller set of types, such as those types whose elements
can be compared for equality. Type classes can be seen as providing a structured way to do just
this. Indeed, we can think of parametric polymorphism as a kind of overloading too! It's just that
the overloading occurs implicitly over all types instead of a constrained set of types (i.e. a type
class).

Comparison to Other Languages. The classes used by Haskell are similar to those used in
other object-oriented languages such as C++ and Java. However, there are some signi�cant di�er-
ences:

� Haskell separates the de�nition of a type from the de�nition of the methods associated with
that type. A class in C++ or Java usually de�nes both a data structure (the member
variables) and the functions associated with the structure (the methods). In Haskell, these
de�nitions are separated.

� The class methods de�ned by a Haskell class correspond to virtual functions in a C++ class.
Each instance of a class provides its own de�nition for each method; class defaults correspond
to default de�nitions for a virtual function in the base class.

� Haskell classes are roughly similar to a Java interface. Like an interface declaration, a Haskell
class declaration de�nes a protocol for using an object rather than de�ning an object itself.

� Haskell does not support the C++ overloading style in which functions with di�erent types
share a common name.

� The type of a Haskell object cannot be implicitly coerced; there is no universal base class
such as Object which values can be projected into or out of.

� C++ and Java attach identifying information (such as a VTable) to the runtime representation
of an object. In Haskell, such information is attached logically instead of physically to values,
through the type system.

� There is no access control (such as public or private class constituents) built into the Haskell
class system. Instead, the module system must be used to hide or reveal components of a
class.

6 Types, Again

Here we examine some of the more advanced aspects of type declarations.

6.1 The Newtype Declaration

A common programming practice is to de�ne a type whose representation is identical to an existing
one but which has a separate identity in the type system. In Haskell, the newtype declaration
creates a new type from an existing one. For example, natural numbers can be represented by the
type Integer using the following declaration:

newtype Natural = MakeNatural Integer

This creates an entirely new type, Natural, whose only constructor contains a single Integer. The
constructor MakeNatural converts between an Natural and an Integer:

toNatural :: Integer -> Natural

toNatural x | x < 0 = error "Can't create negative naturals!"

| otherwise = MakeNatural x

fromNatural :: Natural -> Integer

fromNatural (MakeNatural i) = i

The following instance declaration admits Natural to the Num class:

instance Num Natural where

fromInteger = toNatural

x + y = toNatural (fromNatural x + fromNatural y)

x - y = let r = fromNatural x - fromNatural y in

if r < 0 then error "Unnatural subtraction"

else toNatural r

x * y = toNatural (fromNatural x * fromNatural y)

Without this declaration, Natural would not be in Num. Instances declared for the old type do not
carry over to the new one. Indeed, the whole purpose of this type is to introduce a di�erent Num
instance. This would not be possible if Natural were de�ned as a type synonym of Integer.

All of this works using a data declaration instead of a newtype declaration. However, the data
declaration incurs extra overhead in the representation of Natural values. The use of newtype
avoids the extra level of indirection (caused by laziness) that the data declaration would introduce.

See section 4.2.3 of the report for a more discussion of the relation between newtype, data, and
type declarations.

[Except for the keyword, the newtype declaration uses the same syntax as a data declaration
with a single constructor containing a single �eld. This is appropriate since types de�ned using
newtype are nearly identical to those created by an ordinary data declaration.]

6.2 Field Labels

The �elds within a Haskell data type can be accessed either positionally or by name using �eld labels .
Consider a data type for a two-dimensional point:

data Point = Pt Float Float

The two components of a Point are the �rst and second arguments to the constructor Pt. A
function such as

pointx :: Point -> Float

pointx (Pt x _) = x

may be used to refer to the �rst component of a point in a more descriptive way, but, for large
structures, it becomes tedious to create such functions by hand.

Constructors in a data declaration may be declared with associated �eld names , enclosed in
braces. These �eld names identify the components of constructor by name rather than by position.
This is an alternative way to de�ne Point:

data Point = Pt {pointx, pointy :: Float}

This data type is identical to the earlier de�nition of Point. The constructor Pt is the same in
both cases. However, this declaration also de�nes two �eld names, pointx and pointy. These �eld
names can be used as selector functions to extract a component from a structure. In this example,
the selectors are:

pointx :: Point -> Float

pointy :: Point -> Float

This is a function using these selectors:

absPoint :: Point -> Float

absPoint p = sqrt (pointx p * pointx p +

pointy p * pointy p)

Field labels can also be used to construct new values. The expression Pt {pointx=1, pointy=2}

is identical to Pt 1 2. The use of �eld names in the declaration of a data constructor does not pre-
clude the positional style of �eld access; both Pt {pointx=1, pointy=2} and Pt 1 2 are allowed.
When constructing a value using �eld names, some �elds may be omitted; these absent �elds are
unde�ned.

Pattern matching using �eld names uses a similar syntax for the constructor Pt:

absPoint (Pt {pointx = x, pointy = y}) = sqrt (x*x + y*y)

An update function uses �eld values in an existing structure to �ll in components of a new
structure. If p is a Point, then p {pointx=2} is a point with the same pointy as p but with
pointx replaced by 2. This is not a destructive update: the update function merely creates a new
copy of the object, �lling in the speci�ed �elds with new values.

[The braces used in conjunction with �eld labels are somewhat special: Haskell syntax usually
allows braces to be omitted using the layout rule (described in Section 4.6). However, the braces
associated with �eld names must be explicit.]

Field names are not restricted to types with a single constructor (commonly called `record'
types). In a type with multiple constructors, selection or update operations using �eld names may
fail at runtime. This is similar to the behavior of the head function when applied to an empty list.

Field labels share the top level namespace with ordinary variables and class methods. A �eld
name cannot be used in more than one data type in scope. However, within a data type, the same
�eld name can be used in more than one of the constructors so long as it has the same typing in
all cases. For example, in this data type

data T = C1 {f :: Int, g :: Float}

| C2 {f :: Int, h :: Bool}

the �eld name f applies to both constructors in T. Thus if x is of type T, then x {f=5} will work
for values created by either of the constructors in T.

Field names does not change the basic nature of an algebraic data type; they are simply a
convenient syntax for accessing the components of a data structure by name rather than by position.
They make constructors with many components more manageable since �elds can be added or
removed without changing every reference to the constructor. For full details of �eld labels and
their semantics, see Section x4.2.1.

6.3 Strict Data Constructors

Data structures in Haskell are generally lazy : the components are not evaluated until needed. This
permits structures that contain elements which, if evaluated, would lead to an error or fail to
terminate. Lazy data structures enhance the expressiveness of Haskell and are an essential aspect
of the Haskell programming style.

Internally, each �eld of a lazy data object is wrapped up in a structure commonly referred to
as a thunk that encapsulates the computation de�ning the �eld value. This thunk is not entered
until the value is needed; thunks which contain errors (?) do not a�ect other elements of a data
structure. For example, the tuple ('a',?) is a perfectly legal Haskell value. The 'a' may be
used without disturbing the other component of the tuple. Most programming languages are strict
instead of lazy: that is, all components of a data structure are reduced to values before being placed
in the structure.

There are a number of overheads associated with thunks: they take time to construct and
evaluate, they occupy space in the heap, and they cause the garbage collector to retain other
structures needed for the evaluation of the thunk. To avoid these overheads, strictness 
ags in



30 7 INPUT/OUTPUT

data declarations allow speci�c �elds of a constructor to be evaluated immediately, selectively
suppressing laziness. A �eld marked by ! in a data declaration is evaluated when the structure is
created instead of delayed in a thunk.

There are a number of situations where it may be appropriate to use strictness 
ags:

� Structure components that are sure to be evaluated at some point during program execution.

� Structure components that are simple to evaluate and never cause errors.

� Types in which partially unde�ned values are not meaningful.

For example, the complex number library de�nes the Complex type as:

data RealFloat a => Complex a = !a :+ !a

[note the in�x de�nition of the constructor :+.] This de�nition marks the two components, the
real and imaginary parts, of the complex number as being strict. This is a more compact repre-
sentation of complex numbers but this comes at the expense of making a complex number with an
unde�ned component, 1 :+ ? for example, totally unde�ned (?). As there is no real need for
partially de�ned complex numbers, it makes sense to use strictness 
ags to achieve a more eÆcient
representation.

Strictness 
ags may be used to address memory leaks: structures retained by the garbage
collector but no longer necessary for computation.

The strictness 
ag, !, can only appear in data declarations. It cannot be used in other type
signatures or in any other type de�nitions. There is no corresponding way to mark function
arguments as being strict, although the same e�ect can be obtained using the seq or !$ functions.
See x4.2.1 for further details.

It is diÆcult to present exact guidelines for the use of strictness 
ags. They should be used
with caution: laziness is one of the fundamental properties of Haskell and adding strictness 
ags
may lead to hard to �nd in�nite loops or have other unexpected consequences.

7 Input/Output

The I/O system in Haskell is purely functional, yet has all of the expressive power found in con-
ventional programming languages. In imperative languages, programs proceed via actions which
examine and modify the current state of the world. Typical actions include reading and setting
global variables, writing �les, reading input, and opening windows. Such actions are also a part of
Haskell but are cleanly separated from the purely functional core of the language.

Haskell's I/O system is built around a somewhat daunting mathematical foundation: the
monad . However, understanding of the underlying monad theory is not necessary to program
using the I/O system. Rather, monads are a conceptual structure into which I/O happens to �t. It
is no more necessary to understand monad theory to perform Haskell I/O than it is to understand
group theory to do simple arithmetic. A detailed explanation of monads is found in Section 9.

The monadic operators that the I/O system is built upon are also used for other purposes; we
will look more deeply into monads later. For now, we will avoid the term monad and concentrate
on the use of the I/O system. It's best to think of the I/O monad as simply an abstract data type.

Actions are de�ned rather than invoked within the expression language of Haskell. Evaluating
the de�nition of an action doesn't actually cause the action to happen. Rather, the invocation of
actions takes place outside of the expression evaluation we have considered up to this point.

Actions are either atomic, as de�ned in system primitives, or are a sequential composition of
other actions. The I/O monad contains primitives which build composite actions, a process similar
to joining statements in sequential order using `;' in other languages. Thus the monad serves as
the glue which binds together the actions in a program.

7.1 Basic I/O Operations

Every I/O action returns a value. In the type system, the return value is `tagged' with IO type,
distinguishing actions from other values. For example, the type of the function getChar is:

getChar :: IO Char

The IO Char indicates that getChar, when invoked, performs some action which returns a char-
acter. Actions which return no interesting values use the unit type, (). For example, the putChar
function:

putChar :: Char -> IO ()

takes a character as an argument but returns nothing useful. The unit type is similar to void in
other languages.

Actions are sequenced using an operator that has a rather cryptic name: >>= (or `bind'). Instead
of using this operator directly, we choose some syntactic sugar, the do notation, to hide these
sequencing operators under a syntax resembling more conventional languages. The do notation can
be trivially expanded to >>=, as described in x3.14.

The keyword do introduces a sequence of statements which are executed in order. A statement
is either an action, a pattern bound to the result of an action using <-, or a set of local de�nitions
introduced using let. The do notation uses layout in the same manner as let or where so we can
omit braces and semicolons with proper indentation. Here is a simple program to read and then
print a character:

main :: IO ()

main = do c <- getChar

putChar c

The use of the name main is important: main is de�ned to be the entry point of a Haskell program
(similar to the main function in C), and must have an IO type, usually IO (). (The name main

is special only in the module Main; we will have more to say about modules later.) This program
performs two actions in sequence: �rst it reads in a character, binding the result to the variable
c, and then prints the character. Unlike a let expression where variables are scoped over all
de�nitions, the variables de�ned by <- are only in scope in the following statements.

There is still one missing piece. We can invoke actions and examine their results using do, but
how do we return a value from a sequence of actions? For example, consider the ready function
that reads a character and returns True if the character was a `y':

ready :: IO Bool

ready = do c <- getChar

c == 'y' -- Bad!!!

This doesn't work because the second statement in the `do' is just a boolean value, not an action.
We need to take this boolean and create an action that does nothing but return the boolean as its
result. The return function does just that:

return :: a -> IO a

The return function completes the set of sequencing primitives. The last line of ready should read
return (c == 'y').

We are now ready to look at more complicated I/O functions. First, the function getLine:

getLine :: IO String

getLine = do c <- getChar

if c == '\n'

then return ""

else do l <- getLine

return (c:l)

Note the second do in the else clause. Each do introduces a single chain of statements. Any
intervening construct, such as the if, must use a new do to initiate further sequences of actions.

The return function admits an ordinary value such as a boolean to the realm of I/O actions.
What about the other direction? Can we invoke some I/O actions within an ordinary expression?
For example, how can we say x + print y in an expression so that y is printed out as the expression
evaluates? The answer is that we can't! It is not possible to sneak into the imperative universe
while in the midst of purely functional code. Any value `infected' by the imperative world must be
tagged as such. A function such as

f :: Int -> Int -> Int

absolutely cannot do any I/O since IO does not appear in the returned type. This fact is often
quite distressing to programmers used to placing print statements liberally throughout their code
during debugging. There are, in fact, some unsafe functions available to get around this problem
but these are better left to advanced programmers. Debugging packages (like Trace) often make
liberal use of these `forbidden functions' in an entirely safe manner.

7.2 Programming With Actions

I/O actions are ordinary Haskell values: they may be passed to functions, placed in structures, and
used as any other Haskell value. Consider this list of actions:

todoList :: [IO ()]

todoList = [putChar 'a',

do putChar 'b'

putChar 'c',

do c <- getChar

putChar c]

This list doesn't actually invoke any actions|it simply holds them. To join these actions into a
single action, a function such as sequence_ is needed:

sequence_ :: [IO ()] -> IO ()

sequence_ [] = return ()

sequence_ (a:as) = do a

sequence as

This can be simpli�ed by noting that do x;y is expanded to x >> y (see Section 9.1). This pattern
of recursion is captured by the foldr function (see the Prelude for a de�nition of foldr); a better
de�nition of sequence_ is:

sequence_ :: [IO ()] -> IO ()

sequence_ = foldr (>>) (return ())

The do notation is a useful tool but in this case the underlying monadic operator, >>, is more
appropriate. An understanding of the operators upon which do is built is quite useful to the
Haskell programmer.

The sequence_ function can be used to construct putStr from putChar:

putStr :: String -> IO ()

putStr s = sequence_ (map putChar s)

One of the di�erences between Haskell and conventional imperative programming can be seen in
putStr. In an imperative language, mapping an imperative version of putChar over the string
would be suÆcient to print it. In Haskell, however, the map function does not perform any action.
Instead it creates a list of actions, one for each character in the string. The folding operation in
sequence_ uses the >> function to combine all of the individual actions into a single action. The
return () used here is quite necessary { foldr needs a null action at the end of the chain of
actions it creates (especially if there are no characters in the string!).

The Prelude and the libraries contains many functions which are useful for sequencing I/O
actions. These are usually generalized to arbitrary monads; any function with a context including
Monad m => works with the IO type.

7.3 Exception Handling

So far, we have avoided the issue of exceptions during I/O operations. What would happen if
getChar encounters an end of �le?13 To deal with exceptional conditions such as `�le not found'

13We use the term error for ?: a condition which cannot be recovered from such as non-termination or pattern
match failure. Exceptions, on the other hand, can be caught and handled within the I/O monad.

within the I/O monad, a handling mechanism is used, similar in functionality to the one in standard
ML. No special syntax or semantics are used; exception handling is part of the de�nition of the
I/O sequencing operations.

Errors are encoded using a special data type, IOError. This type represents all possible excep-
tions that may occur within the I/O monad. This is an abstract type: no constructors for IOError
are available to the user. Predicates allow IOError values to be queried. For example, the function

isEOFError :: IOError -> Bool

determines whether an error was caused by an end-of-�le condition. By making IOError abstract,
new sorts of errors may be added to the system without a noticeable change to the data type. The
function isEOFError is de�ned in a separate library, IO, and must be explicitly imported into a
program.

An exception handler has type IOError -> IO a. The catch function associates an exception
handler with an action or set of actions:

catch :: IO a -> (IOError -> IO a) -> IO a

The arguments to catch are an action and a handler. If the action succeeds, its result is returned
without invoking the handler. If an error occurs, it is passed to the handler as a value of type
IOError and the action associated with the handler is then invoked. For example, this version of
getChar returns a newline when an error is encountered:

getChar' :: IO Char

getChar' = getChar `catch` (\e -> return '\n')

This is rather crude since it treats all errors in the same manner. If only end-of-�le is to be
recognized, the error value must be queried:

getChar' :: IO Char

getChar' = getChar `catch` eofHandler where

eofHandler e = if isEofError e then return '\n' else ioError e

The ioError function used here throws an exception on to the next exception handler. The type
of ioError is

ioError :: IOError -> IO a

It is similar to return except that it transfers control to the exception handler instead of proceeding
to the next I/O action. Nested calls to catch are permitted, and produce nested exception handlers.

Using getChar', we can rede�ne getLine to demonstrate the use of nested handlers:

getLine' :: IO String

getLine' = catch getLine'' (\err -> return ("Error: " ++ show err))

where

getLine'' = do c <- getChar'

if c == '\n' then return ""

else do l <- getLine'

return (c:l)

The nested error handlers allow getChar' to catch end of �le while any other error results in a
string starting with "Error: " from getLine'.

For convenience, Haskell provides a default exception handler at the topmost level of a program
that prints out the exception and terminates the program.

7.4 Files, Channels, and Handles

Aside from the I/O monad and the exception handling mechanism it provides, I/O facilities in
Haskell are for the most part quite similar to those in other languages. Many of these functions
are in the IO library instead of the Prelude and thus must be explicitly imported to be in scope
(modules and importing are discussed in Section 11). Also, many of these functions are discussed
in the Library Report instead of the main report.

Opening a �le creates a handle (of type Handle) for use in I/O transactions. Closing the handle
closes the associated �le:

type FilePath = String -- path names in the file system

openFile :: FilePath -> IOMode -> IO Handle

hClose :: Handle -> IO ()

data IOMode = ReadMode | WriteMode | AppendMode | ReadWriteMode

Handles can also be associated with channels : communication ports not directly attached to �les. A
few channel handles are prede�ned, including stdin (standard input), stdout (standard output),
and stderr (standard error). Character level I/O operations include hGetChar and hPutChar,
which take a handle as an argument. The getChar function used previously can be de�ned as:

getChar = hGetChar stdin

Haskell also allows the entire contents of a �le or channel to be returned as a single string:

getContents :: Handle -> IO String

Pragmatically, it may seem that getContents must immediately read an entire �le or channel,
resulting in poor space and time performance under certain conditions. However, this is not the
case. The key point is that getContents returns a \lazy" (i.e. non-strict) list of characters (recall
that strings are just lists of characters in Haskell), whose elements are read \by demand" just like
any other list. An implementation can be expected to implement this demand-driven behavior by
reading one character at a time from the �le as they are required by the computation.

In this example, a Haskell program copies one �le to another:

main = do fromHandle <- getAndOpenFile "Copy from: " ReadMode

toHandle <- getAndOpenFile "Copy to: " WriteMode

contents <- hGetContents fromHandle

hPutStr toHandle contents

hClose toHandle

putStr "Done."

getAndOpenFile :: String -> IOMode -> IO Handle

getAndOpenFile prompt mode =

do putStr prompt

name <- getLine

catch (openFile name mode)

(\_ -> do putStrLn ("Cannot open "++ name ++ "\n")

getAndOpenFile prompt mode)

By using the lazy getContents function, the entire contents of the �le need not be read into
memory all at once. If hPutStr chooses to bu�er the output by writing the string in �xed sized
blocks of characters, only one block of the input �le needs to be in memory at once. The input �le
is closed implicitly when the last character has been read.

7.5 Haskell and Imperative Programming

As a �nal note, I/O programming raises an important issue: this style looks suspiciously like
ordinary imperative programming. For example, the getLine function:

getLine = do c <- getChar

if c == '\n'

then return ""

else do l <- getLine

return (c:l)

bears a striking similarity to imperative code (not in any real language) :

function getLine() {

c := getChar();

if c == `\n` then return ""

else {l := getLine();

return c:l}}

So, in the end, has Haskell simply re-invented the imperative wheel?

In some sense, yes. The I/O monad constitutes a small imperative sub-language inside Haskell,
and thus the I/O component of a program may appear similar to ordinary imperative code. But
there is one important di�erence: There is no special semantics that the user needs to deal with. In
particular, equational reasoning in Haskell is not compromised. The imperative feel of the monadic
code in a program does not detract from the functional aspect of Haskell. An experienced functional
programmer should be able to minimize the imperative component of the program, only using

the I/O monad for a minimal amount of top-level sequencing. The monad cleanly separates the
functional and imperative program components. In contrast, imperative languages with functional
subsets do not generally have any well-de�ned barrier between the purely functional and imperative
worlds.

8 Standard Haskell Classes

In this section we introduce the prede�ned standard type classes in Haskell. We have simpli�ed
these classes somewhat by omitting some of the less interesting methods in these classes; the Haskell
report contains a more complete description. Also, some of the standard classes are part of the
standard Haskell libraries; these are described in the Haskell Library Report.

8.1 Equality and Ordered Classes

The classes Eq and Ord have already been discussed. The de�nition of Ord in the Prelude is
somewhat more complex than the simpli�ed version of Ord presented earlier. In particular, note
the compare method:

data Ordering = EQ | LT | GT

compare :: Ord a => a -> a -> Ordering

The compare method is suÆcient to de�ne all other methods (via defaults) in this class and is the
best way to create Ord instances.

8.2 The Enumeration Class

Class Enum has a set of operations that underlie the syntactic sugar of arithmetic sequences; for
example, the arithmetic sequence expression [1,3..] stands for enumFromThen 1 3 (see x3.10 for
the formal translation). We can now see that arithmetic sequence expressions can be used to
generate lists of any type that is an instance of Enum. This includes not only most numeric types,
but also Char, so that, for instance, ['a'..'z'] denotes the list of lower-case letters in alphabetical
order. Furthermore, user-de�ned enumerated types like Color can easily be given Enum instance
declarations. If so:

[Red .. Violet] ) [Red, Green, Blue, Indigo, Violet]

Note that such a sequence is arithmetic in the sense that the increment between values is constant,
even though the values are not numbers. Most types in Enum can be mapped onto �xed precision
integers; for these, the fromEnum and toEnum convert between Int and a type in Enum.

8.3 The Read and Show Classes

The instances of class Show are those types that can be converted to character strings (typically
for I/O). The class Read provides operations for parsing character strings to obtain the values they
may represent. The simplest function in the class Show is show:

show :: (Show a) => a -> String

Naturally enough, show takes any value of an appropriate type and returns its representation as a
character string (list of characters), as in show (2+2), which results in "4". This is �ne as far as
it goes, but we typically need to produce more complex strings that may have the representations
of many values in them, as in

"The sum of " ++ show x ++ " and " ++ show y ++ " is " ++ show (x+y) ++ "."

and after a while, all that concatenation gets to be a bit ineÆcient. Speci�cally, let's consider a
function to represent the binary trees of Section 2.2.1 as a string, with suitable markings to show
the nesting of subtrees and the separation of left and right branches (provided the element type is
representable as a string):

showTree :: (Show a) => Tree a -> String

showTree (Leaf x) = show x

showTree (Branch l r) = "<" ++ showTree l ++ "|" ++ showTree r ++ ">"

Because (++) has time complexity linear in the length of its left argument, showTree is potentially
quadratic in the size of the tree.

To restore linear complexity, the function shows is provided:

shows :: (Show a) => a -> String -> String

shows takes a printable value and a string and returns that string with the value's representation
concatenated at the front. The second argument serves as a sort of string accumulator, and show

can now be de�ned as shows with the null accumulator. This is the default de�nition of show in
the Show class de�nition:

show x = shows x ""

We can use shows to de�ne a more eÆcient version of showTree, which also has a string accumulator
argument:

showsTree :: (Show a) => Tree a -> String -> String

showsTree (Leaf x) s = shows x s

showsTree (Branch l r) s= '<' : showsTree l ('|' : showsTree r ('>' : s))

This solves our eÆciency problem (showsTree has linear complexity), but the presentation of this
function (and others like it) can be improved. First, let's create a type synonym:

type ShowS = String -> String

This is the type of a function that returns a string representation of something followed by an
accumulator string. Second, we can avoid carrying accumulators around, and also avoid amassing
parentheses at the right end of long constructions, by using functional composition:

showsTree :: (Show a) => Tree a -> ShowS

showsTree (Leaf x) = shows x

showsTree (Branch l r) = ('<':) . showsTree l . ('|':) . showsTree r . ('>':)

Something more important than just tidying up the code has come about by this transformation:
we have raised the presentation from an object level (in this case, strings) to a function level. We
can think of the typing as saying that showsTree maps a tree into a showing function. Functions

like ('<' :) or ("a string" ++) are primitive showing functions, and we build up more complex
functions by function composition.

Now that we can turn trees into strings, let's turn to the inverse problem. The basic idea is
a parser for a type a, which is a function that takes a string and returns a list of (a, String)

pairs [9]. The Prelude provides a type synonym for such functions:

type ReadS a = String -> [(a,String)]

Normally, a parser returns a singleton list, containing a value of type a that was read from the input
string and the remaining string that follows what was parsed. If no parse was possible, however, the
result is the empty list, and if there is more than one possible parse (an ambiguity), the resulting
list contains more than one pair. The standard function reads is a parser for any instance of Read:

reads :: (Read a) => ReadS a

We can use this function to de�ne a parsing function for the string representation of binary trees
produced by showsTree. List comprehensions give us a convenient idiom for constructing such
parsers:14

readsTree :: (Read a) => ReadS (Tree a)

readsTree ('<':s) = [(Branch l r, u) | (l, '|':t) <- readsTree s,

(r, '>':u) <- readsTree t ]

readsTree s = [(Leaf x, t) | (x,t) <- reads s]

Let's take a moment to examine this function de�nition in detail. There are two main cases to
consider: If the �rst character of the string to be parsed is '<', we should have the representation
of a branch; otherwise, we have a leaf. In the �rst case, calling the rest of the input string following
the opening angle bracket s, any possible parse must be a tree Branch l r with remaining string
u, subject to the following conditions:

1. The tree l can be parsed from the beginning of the string s.

2. The string remaining (following the representation of l) begins with '|'. Call the tail of this
string t.

3. The tree r can be parsed from the beginning of t.

4. The string remaining from that parse begins with '>', and u is the tail.

Notice the expressive power we get from the combination of pattern matching with list comprehen-
sion: the form of a resulting parse is given by the main expression of the list comprehension, the
�rst two conditions above are expressed by the �rst generator (\(l, '|':t) is drawn from the list
of parses of s"), and the remaining conditions are expressed by the second generator.

The second de�ning equation above just says that to parse the representation of a leaf, we parse
a representation of the element type of the tree and apply the constructor Leaf to the value thus
obtained.

14An even more elegant approach to parsing uses monads and parser combinators. These are part of a standard
parsing library distributed with most Haskell systems.

We'll accept on faith for the moment that there is a Read (and Show) instance of Integer
(among many other types), providing a reads that behaves as one would expect, e.g.:

(reads "5 golden rings") :: [(Integer,String)] ) [(5, " golden rings")]

With this understanding, the reader should verify the following evaluations:

readsTree "<1|<2|3>>" ) [(Branch (Leaf 1) (Branch (Leaf 2) (Leaf 3)), "")]

readsTree "<1|2" ) []

There are a couple of shortcomings in our de�nition of readsTree. One is that the parser is
quite rigid, allowing no white space before or between the elements of the tree representation; the
other is that the way we parse our punctuation symbols is quite di�erent from the way we parse
leaf values and subtrees, this lack of uniformity making the function de�nition harder to read. We
can address both of these problems by using the lexical analyzer provided by the Prelude:

lex :: ReadS String

lex normally returns a singleton list containing a pair of strings: the �rst lexeme in the input
string and the remainder of the input. The lexical rules are those of Haskell programs, including
comments, which lex skips, along with whitespace. If the input string is empty or contains only
whitespace and comments, lex returns [("","")]; if the input is not empty in this sense, but also
does not begin with a valid lexeme after any leading whitespace and comments, lex returns [].

Using the lexical analyzer, our tree parser now looks like this:

readsTree :: (Read a) => ReadS (Tree a)

readsTree s = [(Branch l r, x) | ("<", t) <- lex s,

(l, u) <- readsTree t,

("|", v) <- lex u,

(r, w) <- readsTree v,

(">", x) <- lex w ]

++

[(Leaf x, t) | (x, t) <- reads s ]

We may now wish to use readsTree and showsTree to declare (Read a) => Tree a an instance
of Read and (Show a) => Tree a an instance of Show. This would allow us to use the generic
overloaded functions from the Prelude to parse and display trees. Moreover, we would automatically
then be able to parse and display many other types containing trees as components, for example,
[Tree Integer]. As it turns out, readsTree and showsTree are of almost the right types to
be Show and Read methods The showsPrec and readsPrec methods are parameterized versions
of shows and reads. The extra parameter is a precedence level, used to properly parenthesize
expressions containing in�x constructors. For types such as Tree, the precedence can be ignored.
The Show and Read instances for Tree are:

instance Show a => Show (Tree a) where

showsPrec _ x = showsTree x

instance Read a => Read (Tree a) where

readsPrec _ s = readsTree s

Alternatively, the Show instance could be de�ned in terms of showTree:

instance Show a => Show (Tree a) where

show t = showTree t

This, however, will be less eÆcient than the ShowS version. Note that the Show class de�nes default
methods for both showsPrec and show, allowing the user to de�ne either one of these in an instance
declaration. Since these defaults are mutually recursive, an instance declaration that de�nes neither
of these functions will loop when called. Other classes such as Num also have these \interlocking
defaults".

We refer the interested reader to xD for details of the Read and Show classes.

We can test the Read and Show instances by applying (read . show) (which should be the
identity) to some trees, where read is a specialization of reads:

read :: (Read a) => String -> a

This function fails if there is not a unique parse or if the input contains anything more than a
representation of one value of type a (and possibly, comments and whitespace).

8.4 Derived Instances

Recall the Eq instance for trees we presented in Section 5; such a declaration is simple|and boring|
to produce: we require that the element type in the leaves be an equality type; then, two leaves are
equal i� they contain equal elements, and two branches are equal i� their left and right subtrees
are equal, respectively. Any other two trees are unequal:

instance (Eq a) => Eq (Tree a) where

(Leaf x) == (Leaf y) = x == y

(Branch l r) == (Branch l' r') = l == l' && r == r'

_ == _ = False

Fortunately, we don't need to go through this tedium every time we need equality operators for
a new type; the Eq instance can be derived automatically from the data declaration if we so specify:

data Tree a = Leaf a | Branch (Tree a) (Tree a) deriving Eq

The deriving clause implicitly produces an Eq instance declaration just like the one in Section 5.
Instances of Ord, Enum, Ix, Read, and Show can also be generated by the deriving clause. [More
than one class name can be speci�ed, in which case the list of names must be parenthesized and
the names separated by commas.]

The derived Ord instance for Tree is slightly more complicated than the Eq instance:

instance (Ord a) => Ord (Tree a) where

(Leaf _) <= (Branch _) = True

(Leaf x) <= (Leaf y) = x <= y

(Branch _) <= (Leaf _) = False

(Branch l r) <= (Branch l' r') = l == l' && r <= r' || l <= l'

This speci�es a lexicographic order: Constructors are ordered by the order of their appearance in

the data declaration, and the arguments of a constructor are compared from left to right. Recall
that the built-in list type is semantically equivalent to an ordinary two-constructor type. In fact,
this is the full declaration:

data [a] = [] | a : [a] deriving (Eq, Ord) -- pseudo-code

(Lists also have Show and Read instances, which are not derived.) The derived Eq and Ord instances
for lists are the usual ones; in particular, character strings, as lists of characters, are ordered as
determined by the underlying Char type, with an initial substring comparing less than a longer
string; for example, "cat" < "catalog".

In practice, Eq and Ord instances are almost always derived, rather than user-de�ned. In fact, we
should provide our own de�nitions of equality and ordering predicates only with some trepidation,
being careful to maintain the expected algebraic properties of equivalence relations and total orders.
An intransitive (==) predicate, for example, could be disastrous, confusing readers of the program
and confounding manual or automatic program transformations that rely on the (==) predicate's
being an approximation to de�nitional equality. Nevertheless, it is sometimes necessary to provide
Eq or Ord instances di�erent from those that would be derived; probably the most important
example is that of an abstract data type in which di�erent concrete values may represent the same
abstract value.

An enumerated type can have a derived Enum instance, and here again, the ordering is that of
the constructors in the data declaration. For example:

data Day = Sunday | Monday | Tuesday | Wednesday

| Thursday | Friday | Saturday deriving (Enum)

Here are some simple examples using the derived instances for this type:

[Wednesday .. Friday] ) [Wednesday, Thursday, Friday]

[Monday, Wednesday ..] ) [Monday, Wednesday, Friday]

Derived Read (Show) instances are possible for all types whose component types also have
Read (Show) instances. (Read and Show instances for most of the standard types are provided
by the Prelude. Some types, such as the function type (->), have a Show instance but not a
corresponding Read.) The textual representation de�ned by a derived Show instance is consistent
with the appearance of constant Haskell expressions of the type in question. For example, if we
add Show and Read to the deriving clause for type Day, above, we obtain

show [Monday .. Wednesday] ) "[Monday,Tuesday,Wednesday]"

9 About Monads

Many newcomers to Haskell are puzzled by the concept of monads. Monads are frequently encoun-
tered in Haskell: the IO system is constructed using a monad, a special syntax for monads has
been provided (do expressions), and the standard libraries contain an entire module dedicated to
monads. In this section we explore monadic programming in more detail.

This section is perhaps less \gentle" than the others. Here we address not only the language
features that involve monads but also try to reveal the bigger picture: why monads are such an
important tool and how they are used. There is no single way of explaining monads that works for
everyone; more explanations can be found at haskell.org. Another good introduction to practical
programming using monads is Wadler's Monads for Functional Programming [10].

9.1 Monadic Classes

The Prelude contains a number of classes de�ning monads are they are used in Haskell. These classes
are based on the monad construct in category theory; whilst the category theoretic terminology
provides the names for the monadic classes and operations, it is not necessary to delve into abstract
mathematics to get an intuitive understanding of how to use the monadic classes.

A monad is constructed on top of a polymorphic type such as IO. The monad itself is de�ned
by instance declarations associating the type with the some or all of the monadic classes, Functor,
Monad, and MonadPlus. None of the monadic classes are derivable. In addition to IO, two other
types in the Prelude are members of the monadic classes: lists ([]) and Maybe.

Mathematically, monads are governed by set of laws that should hold for the monadic operations.
This idea of laws is not unique to monads: Haskell includes other operations that are governed, at
least informally, by laws. For example, x /= y and not (x == y) ought to be the same for any
type of values being compared. However, there is no guarantee of this: both == and /= are separate
methods in the Eq class and there is no way to assure that == and =/ are related in this manner. In
the same sense, the monadic laws presented here are not enforced by Haskell, but ought be obeyed
by any instances of a monadic class. The monad laws give insight into the underlying structure of
monads: by examining these laws, we hope to give a feel for how monads are used.

The Functor class, already discussed in section 5, de�nes a single operation: fmap. The map
function applies an operation to the objects inside a container (polymorphic types can be thought
of as containers for values of another type), returning a container of the same shape. These laws
apply to fmap in the class Functor:

fmap id = id

fmap (f . g) = fmap f . fmap g

These laws ensure that the container shape is unchanged by fmap and that the contents of the
container are not re-arranged by the mapping operation.

The Monad class de�nes two basic operators: >>= (bind) and return.

infixl 1 >>, >>=

class Monad m where

(>>=) :: m a -> (a -> m b) -> m b

(>>) :: m a -> m b -> m b

return :: a -> m a

fail :: String -> m a

m >> k = m >>= \_ -> k

The bind operations, >> and >>=, combine two monadic values while the return operation injects

a value into the monad (container). The signature of >>= helps us to understand this operation:
ma >>= \v -> mb combines a monadic value ma containing values of type a and a function which
operates on a value v of type a, returning the monadic value mb. The result is to combine ma and
mb into a monadic value containing b. The >> function is used when the function does not need
the value produced by the �rst monadic operator.

The precise meaning of binding depends, of course, on the monad. For example, in the IO
monad, x >>= y performs two actions sequentially, passing the result of the �rst into the second.
For the other built-in monads, lists and the Maybe type, these monadic operations can be understood
in terms of passing zero or more values from one calculation to the next. We will see examples of
this shortly.

The do syntax provides a simple shorthand for chains of monadic operations. The essential
translation of do is captured in the following two rules:

do e1 ; e2 = e1 >> e2

do p <- e1; e2 = e1 >>= \p -> e2

When the pattern in this second form of do is refutable, pattern match failure calls the fail

operation. This may raise an error (as in the IO monad) or return a \zero" (as in the list monad).
Thus the more complex translation is

do p <- e1; e2 = e1 >>= (\v -> case v of p -> e2; _ -> fail "s")

where "s" is a string identifying the location of the do statement for possible use in an error
message. For example, in the I/O monad, an action such as 'a' <- getChar will call fail if the
character typed is not 'a'. This, in turn, terminates the program since in the I/O monad fail calls
error.

The laws which govern >>= and return are:

return a >>= k = k a

m >>= return = m

xs >>= return . f = fmap f xs

m >>= (\x -> k x >>= h) = (m >>= k) >>= h

The class MonadPlus is used for monads that have a zero element and a plus operation:

class (Monad m) => MonadPlus m where

mzero :: m a

mplus :: m a -> m a -> m a

The zero element obeys the following laws:

m >>= \x -> mzero = mzero

mzero >>= m = mzero

For lists, the zero value is [], the empty list. The I/O monad has no zero element and is not a
member of this class.

The laws governing the mplus operator are as follows:

m `mplus` mzero = m

mzero `mplus` m = m

The mplus operator is ordinary list concatenation in the list monad.

9.2 Built-in Monads

Given the monadic operations and the laws that govern them, what can we build? We have already
examined the I/O monad in detail so we start with the two other built-in monads.

For lists, monadic binding involves joining together a set of calculations for each value in the
list. When used with lists, the signature of >>= becomes:

(>>=) :: [a] -> (a -> [b]) -> [b]

That is, given a list of a's and a function that maps an a onto a list of b's, binding applies this
function to each of the a's in the input and returns all of the generated b's concatenated into a
list. The return function creates a singleton list. These operations should already be familiar:
list comprehensions can easily be expressed using the monadic operations de�ned for lists. These
following three expressions are all di�erent syntax for the same thing:

[(x,y) | x <- [1,2,3] , y <- [1,2,3], x /= y]

do x <- [1,2,3]

y <- [1,2,3]

True <- return (x /= y)

return (x,y)

[1,2,3] >>= (\ x -> [1,2,3] >>= (\y -> return (x/=y) >>=

(\r -> case r of True -> return (x,y)

_ -> fail "")))

This de�nition depends on the de�nition of fail in this monad as the empty list. Essentially, each
<- is generating a set of values which is passed on into the remainder of the monadic computation.
Thus x <- [1,2,3] invokes the remainder of the monadic computation three times, once for each
element of the list. The returned expression, (x,y), will be evaluated for all possible combinations
of bindings that surround it. In this sense, the list monad can be thought of as describing functions
of multi-valued arguments. For example, this function:

mvLift2 :: (a -> b -> c) -> [a] -> [b] -> [c]

mvLift2 f x y = do x' <- x

y' <- y

return (f x' y')

turns an ordinary function of two arguments (f) into a function over multiple values (lists of
arguments), returning a value for each possible combination of the two input arguments. For
example,

mvLift2 (+) [1,3] [10,20,30] ) [11,21,31,13,23,33]

mvLift2 (\a b->[a,b]) "ab" "cd" ) ["ac","ad","bc","bd"]

mvLift2 (*) [1,2,4] [] ) []

This function is a specialized version of the LiftM2 function in the monad library. You can think
of it as transporting a function from outside the list monad, f, into the list monad in which
computations take on multiple values.

The monad de�ned for Maybe is similar to the list monad: the value Nothing serves as [] and
Just x as [x].

9.3 Using Monads

Explaining the monadic operators and their associated laws doesn't really show what monads are
good for. What they really provide is modularity. That is, by de�ning an operation monadically,
we can hide underlying machinery in a way that allows new features to be incorporated into the
monad transparently. Wadler's paper [10] is an excellent example of how monads can be used to
construct modular programs. We will start with a monad taken directly from this paper, the state
monad, and then build a more complex monad with a similar de�nition.

Brie
y, a state monad built around a state type S looks like this:

data SM a = SM (S -> (a,S)) -- The monadic type

instance Monad SM where

-- defines state propagation

SM c1 >>= fc2 = SM (\s0 -> let (r,s1) = c1 s0

SM c2 = fc2 r in

c2 s1)

return k = SM (\s -> (k,s))

-- extracts the state from the monad

readSM :: SM S

readSM = SM (\s -> (s,s))

-- updates the state of the monad

updateSM :: (S -> S) -> SM () -- alters the state

updateSM f = SM (\s -> ((), f s))

-- run a computation in the SM monad

runSM :: S -> SM a -> (a,S)

runSM s0 (SM c) = c s0

This example de�nes a new type, SM, to be a computation that implicitly carries a type S. That
is, a computation of type SM t de�nes a value of type t while also interacting with (reading and
writing) the state of type S. The de�nition of SM is simple: it consists of functions that take a
state and produce two results: a returned value (of any type) and an updated state. We can't use
a type synonym here: we need a type name like SM that can be used in instance declarations. The
newtype declaration is often used here instead of data.

This instance declaration de�nes the `plumbing' of the monad: how to sequence two com-
putations and the de�nition of an empty computation. Sequencing (the >>= operator) de�nes a
computation (denoted by the constructor SM) that passes an initial state, s0, into c1, then passes
the value coming out of this computation, r, to the function that returns the second computation,
c2. Finally, the state coming out of c1 is passed into c2 and the overall result is the result of c2.

The de�nition of return is easier: return doesn't change the state at all; it only serves to bring
a value into the monad.

While >>= and return are the basic monadic sequencing operations, we also need some monadic
primitives. A monadic primitive is simply an operation that uses the insides of the monad abstrac-
tion and taps into the `wheels and gears' that make the monad work. For example, in the IOmonad,

operators such as putChar are primitive since they deal with the inner workings of the IO monad.
Similarly, our state monad uses two primitives: readSM and updateSM. Note that these depend on
the inner structure of the monad - a change to the de�nition of the SM type would require a change
to these primitives.

The de�nition of readSM and updateSM are simple: readSM brings the state out of the monad
for observation while updateSM allows the user to alter the state in the monad. (We could also
have used writeSM as a primitive but update is often a more natural way of dealing with state).

Finally, we need a function that runs computations in the monad, runSM. This takes an initial
state and a computation and yields both the returned value of the computation and the �nal state.

Looking at the bigger picture, what we are trying to do is de�ne an overall computation as
a series of steps (functions with type SM a), sequenced using >>= and return. These steps may
interact with the state (via readSM or updateSM) or may ignore the state. However, the use (or
non-use) of the state is hidden: we don't invoke or sequence our computations di�erently depending
on whether or not they use S.

Rather than present any examples using this simple state monad, we proceed on to a more
complex example that includes the state monad. We de�ne a small embedded language of resource-
using calculations. That is, we build a special purpose language implemented as a set of Haskell
types and functions. Such languages use the basic tools of Haskell, functions and types, to build a
library of operations and types speci�cally tailored to a domain of interest.

In this example, consider a computation that requires some sort of resource. If the resource is
available, computation proceeds; when the resource is unavailable, the computation suspends. We
use the type R to denote a computation using resources controlled by our monad. The de�nition
of R is as follows:

data R a = R (Resource -> (Resource, Either a (R a)))

Each computation is a function from available resources to remaining resources, coupled with either
a result, of type a, or a suspended computation, of type R a, capturing the work done up to the
point where resources were exhausted.

The Monad instance for R is as follows:

instance Monad R where

R c1 >>= fc2 = R (\r -> case c1 r of

(r', Left v) -> let R c2 = fc2 v in

c2 r'

(r', Right pc1) -> (r', Right (pc1 >>= fc2)))

return v = R (\r -> (r, (Left v)))

The Resource type is used in the same manner as the state in the state monad. This de�nition
reads as follows: to combine two `resourceful' computations, c1 and fc2 (a function producing c2),
pass the initial resources into c1. The result will be either

� a value, v, and remaining resources, which are used to determine the next computation (the
call fc2 v), or

� a suspended computation, pc1, and resources remaining at the point of suspension.

The suspension must take the second computation into consideration: pc1 suspends only the �rst
computation, c1, so we must bind c2 to this to produce a suspension of the overall computation.
The de�nition of return leaves the resources unchanged while moving v into the monad.

This instance declaration de�nes the basic structure of the monad but does not determine how
resources are used. This monad could be used to control many types of resource or implement many
di�erent types of resource usage policies. We will demonstrate a very simple de�nition of resources
as an example: we choose Resource to be an Integer, representing available computation steps:

type Resource = Integer

This function takes a step unless no steps are available:

step :: a -> R a

step v = c where

c = R (\r -> if r /= 0 then (r-1, Left v)

else (r, Right c))

The Left and Right constructors are part of the Either type. This function continues computation
in R by returning v so long as there is at least one computational step resource available. If no steps
are available, the step function suspends the current computation (this suspension is captured in
c) and passes this suspended computation back into the monad.

So far, we have the tools to de�ne a sequence of \resourceful" computations (the monad) and
we can express a form of resource usage using step. Finally, we need to address how computations
in this monad are expressed.

Consider an increment function in our monad:

inc :: R Integer -> R Integer

inc i = do iValue <- i

step (iValue+1)

This de�nes increment as a single step of computation. The <- is necessary to pull the argument
value out of the monad; the type of iValue is Integer instead of R Integer.

This de�nition isn't particularly satisfying, though, compared to the standard de�nition of the
increment function. Can we instead \dress up" existing operations like + so that they work in our
monadic world? We'll start with a set of lifting functions. These bring existing functionality
into the monad. Consider the de�nition of lift1 (this is slightly di�erent from the liftM1 found
in the Monad library):

lift1 :: (a -> b) -> (R a -> R b)

lift1 f = \ra1 -> do a1 <- ra1

step (f a1)

This takes a function of a single argument, f, and creates a function in R that executes the lifted
function in a single step. Using lift1, inc becomes

inc :: R Integer -> R Integer

inc i = lift1 (i+1)

This is better but still not ideal. First, we add lift2:

lift2 :: (a -> b -> c) -> (R a -> R b -> R c)

lift2 f = \ra1 ra2 -> do a1 <- ra1

a2 <- ra2

step (f a1 a2)

Notice that this function explicitly sets the order of evaluation in the lifted function: the compu-
tation yielding a1 occurs before the computation for a2.

Using lift2, we can create a new version of == in the R monad:

(==*) :: Ord a => R a -> R a -> R Bool

(==*) = lift2 (==)

We had to use a slightly di�erent name for this new function since == is already taken but in some
cases we can use the same name for the lifted and unlifted function. This instance declaration
allows all of the operators in Num to be used in R:

instance Num a => Num (R a) where

(+) = lift2 (+)

(-) = lift2 (-)

negate = lift1 negate

(*) = lift2 (*)

abs = lift1 abs

fromInteger = return . fromInteger

The fromInteger function is applied implicitly to all integer constants in a Haskell program (see
Section 10.3); this de�nition allows integer constants to have the type R Integer. We can now,
�nally, write increment in a completely natural style:

inc :: R Integer -> R Integer

inc x = x + 1

Note that we cannot lift the Eq class in the same manner as the Num class: the signature of ==* is
not compatible with allowable overloadings of == since the result of ==* is R Bool instead of Bool.

To express interesting computations in R we will need a conditional. Since we can't use if (it
requires that the test be of type Bool instead of R Bool), we name the function ifR:

ifR :: R Bool -> R a -> R a -> R a

ifR tst thn els = do t <- tst

if t then thn else els

Now we're ready for a larger program in the R monad:

fact :: R Integer -> R Integer

fact x = ifR (x ==* 0) 1 (x * fact (x-1))

Now this isn't quite the same as an ordinary factorial function but still quite readable. The idea
of providing new de�nitions for existing operations like + or if is an essential part of creating an
embedded language in Haskell. Monads are particularly useful for encapsulating the semantics of
these embedded languages in a clean and modular way.

We're now ready to actually run some programs. This function runs a program in M given a
maximum number of computation steps:

run :: Resource -> R a -> Maybe a

run s (R p) = case (p s) of

(_, Left v) -> Just v

_ -> Nothing

We use the Maybe type to deal with the possibility of the computation not �nishing in the allotted
number of steps. We can now compute

run 10 (fact 2) ) Just 2

run 10 (fact 20) ) Nothing

Finally, we can add some more interesting functionality to this monad. Consider the following
function:

(|||) :: R a -> R a -> R a

This runs two computations in parallel, returning the value of the �rst one to complete. One
possible de�nition of this function is:

c1 ||| c2 = oneStep c1 (\c1' -> c2 ||| c1')

where

oneStep :: R a -> (R a -> R a) -> R a

oneStep (R c1) f =

R (\r -> case c1 1 of

(r', Left v) -> (r+r'-1, Left v)

(r', Right c1') -> -- r' must be 0

let R next = f c1' in

next (r+r'-1))

This takes a step in c1, returning its value of c1 complete or, if c1 returns a suspended computation
(c1'), it evaluates c2 ||| c1'. The oneStep function takes a single step in its argument, either
returning an evaluated value or passing the remainder of the computation into f. The de�nition
of oneStep is simple: it gives c1 a 1 as its resource argument. If a �nal value is reached, this is
returned, adjusting the returned step count (it is possible that a computation might return after
taking no steps so the returned resource count isn't necessarily 0). If the computation suspends, a
patched up resource count is passed to the �nal continuation.

We can now evaluate expressions like run 100 (fact (-1) ||| (fact 3)) without looping
since the two calculations are interleaved. (Our de�nition of fact loops for -1). Many variations
are possible on this basic structure. For example, we could extend the state to include a trace of
the computation steps. We could also embed this monad inside the standard IO monad, allowing
computations in M to interact with the outside world.

While this example is perhaps more advanced than others in this tutorial, it serves to illustrate
the power of monads as a tool for de�ning the basic semantics of a system. We also present
this example as a model of a small Domain Speci�c Language, something Haskell is particularly
good at de�ning. Many other DSLs have been developed in Haskell; see haskell.org for many
more examples. Of particular interest are Fran, a language of reactive animations, and Haskore, a
language of computer music.

10 Numbers

Haskell provides a rich collection of numeric types, based on those of Scheme [7], which in turn
are based on Common Lisp [8]. (Those languages, however, are dynamically typed.) The standard
types include �xed- and arbitrary-precision integers, ratios (rational numbers) formed from each
integer type, and single- and double-precision real and complex 
oating-point. We outline here the
basic characteristics of the numeric type class structure and refer the reader to x6.4 for details.

10.1 Numeric Class Structure

The numeric type classes (class Num and those that lie below it) account for many of the standard
Haskell classes. We also note that Num is a subclass of Eq, but not of Ord; this is because the order
predicates do not apply to complex numbers. The subclass Real of Num, however, is a subclass of
Ord as well.

The Num class provides several basic operations common to all numeric types; these include,
among others, addition, subtraction, negation, multiplication, and absolute value:

(+), (-), (*) :: (Num a) => a -> a -> a

negate, abs :: (Num a) => a -> a

[negate is the function applied by Haskell's only pre�x operator, minus; we can't call it (-), because
that is the subtraction function, so this name is provided instead. For example, -x*y is equivalent
to negate (x*y). (Pre�x minus has the same syntactic precedence as in�x minus, which, of course,
is lower than that of multiplication.)]

Note that Num does not provide a division operator; two di�erent kinds of division operators are
provided in two non-overlapping subclasses of Num:

The class Integral provides whole-number division and remainder operations. The standard
instances of Integral are Integer (unbounded or mathematical integers, also known as \bignums")
and Int (bounded, machine integers, with a range equivalent to at least 29-bit signed binary). A
particular Haskell implementation might provide other integral types in addition to these. Note
that Integral is a subclass of Real, rather than of Num directly; this means that there is no attempt
to provide Gaussian integers.

All other numeric types fall in the class Fractional, which provides the ordinary division
operator (/). The further subclass Floating contains trigonometric, logarithmic, and exponential
functions.

The RealFrac subclass of Fractional and Real provides a function properFraction, which
decomposes a number into its whole and fractional parts, and a collection of functions that round
to integral values by di�ering rules:

properFraction :: (Fractional a, Integral b) => a -> (b,a)

truncate, round,

floor, ceiling: :: (Fractional a, Integral b) => a -> b

The RealFloat subclass of Floating and RealFrac provides some specialized functions for
eÆcient access to the components of a 
oating-point number, the exponent and signi�cand. The
standard types Float and Double fall in class RealFloat.

10.2 Constructed Numbers

Of the standard numeric types, Int, Integer, Float, and Double are primitive. The others are
made from these by type constructors.

Complex (found in the library Complex) is a type constructor that makes a complex type in
class Floating from a RealFloat type:

data (RealFloat a) => Complex a = !a :+ !a deriving (Eq, Text)

The ! symbols are strictness 
ags; these were discussed in Section 6.3. Notice the context
RealFloat a, which restricts the argument type; thus, the standard complex types are Complex Float

and Complex Double. We can also see from the data declaration that a complex number is written
x :+ y ; the arguments are the cartesian real and imaginary parts, respectively. Since :+ is a data
constructor, we can use it in pattern matching:

conjugate :: (RealFloat a) => Complex a -> Complex a

conjugate (x:+y) = x :+ (-y)

Similarly, the type constructor Ratio (found in the Rational library) makes a rational type in
class RealFrac from an instance of Integral. (Rational is a type synonym for Ratio Integer.)
Ratio, however, is an abstract type constructor. Instead of a data constructor like :+, rationals
use the `%' function to form a ratio from two integers. Instead of pattern matching, component
extraction functions are provided:

(%) :: (Integral a) => a -> a -> Ratio a

numerator, denominator :: (Integral a) => Ratio a -> a

Why the di�erence? Complex numbers in cartesian form are unique|there are no nontrivial
identities involving :+. On the other hand, ratios are not unique, but have a canonical (reduced)
form that the implementation of the abstract data type must maintain; it is not necessarily the
case, for instance, that numerator (x%y) is equal to x, although the real part of x:+y is always x.

10.3 Numeric Coercions and Overloaded Literals

The Standard Prelude and libraries provide several overloaded functions that serve as explicit
coercions:

fromInteger :: (Num a) => Integer -> a

fromRational :: (Fractional a) => Rational -> a

toInteger :: (Integral a) => a -> Integer

toRational :: (RealFrac a) => a -> Rational

fromIntegral :: (Integral a, Num b) => a -> b

fromRealFrac :: (RealFrac a, Fractional b) => a -> b

fromIntegral = fromInteger . toInteger

fromRealFrac = fromRational . toRational

Two of these are implicitly used to provide overloaded numeric literals: An integer numeral
(without a decimal point) is actually equivalent to an application of fromInteger to the value of
the numeral as an Integer. Similarly, a 
oating numeral (with a decimal point) is regarded as
an application of fromRational to the value of the numeral as a Rational. Thus, 7 has the type
(Num a) => a, and 7.3 has the type (Fractional a) => a. This means that we can use numeric
literals in generic numeric functions, for example:

halve :: (Fractional a) => a -> a

halve x = x * 0.5

This rather indirect way of overloading numerals has the additional advantage that the method of
interpreting a numeral as a number of a given type can be speci�ed in an Integral or Fractional
instance declaration (since fromInteger and fromRational are operators of those classes, respec-
tively). For example, the Num instance of (RealFloat a) => Complex a contains this method:

fromInteger x = fromInteger x :+ 0

This says that a Complex instance of fromInteger is de�ned to produce a complex number whose
real part is supplied by an appropriate RealFloat instance of fromInteger. In this manner, even
user-de�ned numeric types (say, quaternions) can make use of overloaded numerals.

As another example, recall our �rst de�nition of inc from Section 2:

inc :: Integer -> Integer

inc n = n+1

Ignoring the type signature, the most general type of inc is (Num a) => a->a. The explicit type
signature is legal, however, since it is more speci�c than the principal type (a more general type
signature would cause a static error). The type signature has the e�ect of restricting inc's type,
and in this case would cause something like inc (1::Float) to be ill-typed.

10.4 Default Numeric Types

Consider the following function de�nition:

rms :: (Floating a) => a -> a -> a

rms x y = sqrt ((x^2 + y^2) * 0.5)

The exponentiation function (^) (one of three di�erent standard exponentiation operators with
di�erent typings, see x6.8.5) has the type (Num a, Integral b) => a -> b -> a, and since 2 has
the type (Num a) => a, the type of x^2 is (Num a, Integral b) => a. This is a problem; there

is no way to resolve the overloading associated with the type variable b, since it is in the context,
but has otherwise vanished from the type expression. Essentially, the programmer has speci�ed
that x should be squared, but has not speci�ed whether it should be squared with an Int or an
Integer value of two. Of course, we can �x this:

rms x y = sqrt ((x ^ (2::Integer) + y ^ (2::Integer)) * 0.5)

It's obvious that this sort of thing will soon grow tiresome, however.

In fact, this kind of overloading ambiguity is not restricted to numbers:

show (read "xyz")

As what type is the string supposed to be read? This is more serious than the exponentiation
ambiguity, because there, any Integral instance will do, whereas here, very di�erent behavior can
be expected depending on what instance of Text is used to resolve the ambiguity.

Because of the di�erence between the numeric and general cases of the overloading ambiguity
problem, Haskell provides a solution that is restricted to numbers: Each module may contain
a default declaration, consisting of the keyword default followed by a parenthesized, comma-
separated list of numeric monotypes (types with no variables). When an ambiguous type variable
is discovered (such as b, above), if at least one of its classes is numeric and all of its classes are
standard, the default list is consulted, and the �rst type from the list that will satisfy the context
of the type variable is used. For example, if the default declaration default (Int, Float) is in
e�ect, the ambiguous exponent above will be resolved as type Int. (See x4.3.4 for more details.)

The \default default" is (Integer, Double), but (Integer, Rational, Double) may also be
appropriate. Very cautious programmers may prefer default (), which provides no defaults.

11 Modules

A Haskell program consists of a collection of modules. A module in Haskell serves the dual purpose
of controlling name-spaces and creating abstract data types.

The top level of a module contains any of the various declarations we have discussed: �xity
declarations, data and type declarations, class and instance declarations, type signatures, function
de�nitions, and pattern bindings. Except for the fact that import declarations (to be described
shortly) must appear �rst, the declarations may appear in any order (the top-level scope is mutually
recursive).

Haskell's module design is relatively conservative: the name-space of modules is completely 
at,
and modules are in no way \�rst-class." Module names are alphanumeric and must begin with an
uppercase letter. There is no formal connection between a Haskell module and the �le system that
would (typically) support it. In particular, there is no connection between module names and �le
names, and more than one module could conceivably reside in a single �le (one module may even
span several �les). Of course, a particular implementation will most likely adopt conventions that
make the connection between modules and �les more stringent.

Technically speaking, a module is really just one big declaration which begins with the keyword
module; here's an example for a module whose name is Tree:

module Tree ( Tree(Leaf,Branch), fringe ) where

data Tree a = Leaf a | Branch (Tree a) (Tree a)

fringe :: Tree a -> [a]

fringe (Leaf x) = [x]

fringe (Branch left right) = fringe left ++ fringe right

The type Tree and the function fringe should be familiar; they were given as examples in Section
2.2.1. [Because of the where keyword, layout is active at the top level of a module, and thus the
declarations must all line up in the same column (typically the �rst). Also note that the module
name is the same as that of the type; this is allowed.]

This module explicitly exports Tree, Leaf, Branch, and fringe. If the export list following the
module keyword is omitted, all of the names bound at the top level of the module would be exported.
(In the above example everything is explicitly exported, so the e�ect would be the same.) Note that
the name of a type and its constructors have be grouped together, as in Tree(Leaf,Branch). As
short-hand, we could also write Tree(..). Exporting a subset of the constructors is also possible.
The names in an export list need not be local to the exporting module; any name in scope may be
listed in an export list.

The Tree module may now be imported into some other module:

module Main (main) where

import Tree ( Tree(Leaf,Branch), fringe )

main = print (fringe (Branch (Leaf 1) (Leaf 2)))

The various items being imported into and exported out of a module are called entities. Note the
explicit import list in the import declaration; omitting it would cause all entities exported from
Tree to be imported.

11.1 Quali�ed Names

There is an obvious problem with importing names directly into the namespace of module. What if
two imported modules contain di�erent entities with the same name? Haskell solves this problem
using quali�ed names . An import declaration may use the qualified keyword to cause the im-
ported names to be pre�xed by the name of the module imported. These pre�xes are followed by
the `.' character without intervening whitespace. [Quali�ers are part of the lexical syntax. Thus,
A.x and A . x are quite di�erent: the �rst is a quali�ed name and the second a use of the in�x `.'
function.] For example, using the Tree module introduced above:

module Fringe(fringe) where

import Tree(Tree(..))

fringe :: Tree a -> [a] -- A different definition of fringe

fringe (Leaf x) = [x]

fringe (Branch x y) = fringe x

module Main where

import Tree ( Tree(Leaf,Branch), fringe )

import qualified Fringe ( fringe )

main = do print (fringe (Branch (Leaf 1) (Leaf 2)))

print (Fringe.fringe (Branch (Leaf 1) (Leaf 2)))

Some Haskell programmers prefer to use quali�ers for all imported entities, making the source
of each name explicit with every use. Others prefer short names and only use quali�ers when
absolutely necessary.

Quali�ers are used to resolve con
icts between di�erent entities which have the same name. But
what if the same entity is imported from more than one module? Fortunately, such name clashes
are allowed: an entity can be imported by various routes without con
ict. The compiler knows
whether entities from di�erent modules are actually the same.

11.2 Abstract Data Types

Aside from controlling namespaces, modules provide the only way to build abstract data types
(ADTs) in Haskell. For example, the characteristic feature of an ADT is that the representation
type is hidden; all operations on the ADT are done at an abstract level which does not depend
on the representation. For example, although the Tree type is simple enough that we might not
normally make it abstract, a suitable ADT for it might include the following operations:

data Tree a -- just the type name

leaf :: a -> Tree a

branch :: Tree a -> Tree a -> Tree a

cell :: Tree a -> a

left, right :: Tree a -> Tree a

isLeaf :: Tree a -> Bool

A module supporting this is:

module TreeADT (Tree, leaf, branch, cell,

left, right, isLeaf) where

data Tree a = Leaf a | Branch (Tree a) (Tree a)

leaf = Leaf

branch = Branch

cell (Leaf a) = a

left (Branch l r) = l

right (Branch l r) = r

isLeaf (Leaf _) = True

isLeaf _ = False

Note in the export list that the type name Tree appears alone (i.e. without its constructors). Thus
Leaf and Branch are not exported, and the only way to build or take apart trees outside of the
module is by using the various (abstract) operations. Of course, the advantage of this information
hiding is that at a later time we could change the representation type without a�ecting users of
the type.

11.3 More Features

Here is a brief overview of some other aspects of the module system. See the report for more details.

� An import declaration may selectively hide entities using a hiding clause in the import
declaration. This is useful for explicitly excluding names that are used for other purposes
without having to use quali�ers for other imported names from the module.

� An import may contain an as clause to specify a di�erent quali�er than the name of the
importing module. This can be used to shorten quali�ers from modules with long names or
to easily adapt to a change in module name without changing all quali�ers.

� Programs implicitly import the Prelude module. An explicit import of the Prelude overrides
the implicit import of all Prelude names. Thus,

import Prelude hiding length

will not import length from the Standard Prelude, allowing the name length to be de�ned
di�erently.

� Instance declarations are not explicitly named in import or export lists. Every module exports
all of its instance declarations and every import brings all instance declarations into scope.

� Class methods may be named either in the manner of data constructors, in parentheses
following the class name, or as ordinary variables.

Although Haskell's module system is relatively conservative, there are many rules concerning the
import and export of values. Most of these are obvious|for instance, it is illegal to import two
di�erent entities having the same name into the same scope. Other rules are not so obvious|for
example, for a given type and class, there cannot be more than one instance declaration for that
combination of type and class anywhere in the program. The reader should read the Report for
details (x5).

12 Typing Pitfalls

This short section give an intuitive description of a few common problems that novices run into
using Haskell's type system.

12.1 Let-Bound Polymorphism

Any language using the Hindley-Milner type system has what is called let-bound polymorphism,
because identi�ers not bound using a let or where clause (or at the top level of a module) are
limited with respect to their polymorphism. In particular, a lambda-bound function (i.e., one passed
as argument to another function) cannot be instantiated in two di�erent ways. For example, this
program is illegal:

let f g = (g [], g 'a') -- ill-typed expression

in f (\x->x)

because g, bound to a lambda abstraction whose principal type is a->a, is used within f in two
di�erent ways: once with type [a]->[a], and once with type Char->Char.

12.2 Numeric Overloading

It is easy to forget at times that numerals are overloaded, and not implicitly coerced to the various
numeric types, as in many other languages. More general numeric expressions sometimes cannot
be quite so generic. A common numeric typing error is something like the following:

average xs = sum xs / length xs -- Wrong!

(/) requires fractional arguments, but length's result is an Int. The type mismatch must be
corrected with an explicit coercion:

average :: (Fractional a) => [a] -> a

average xs = sum xs / fromIntegral (length xs)

12.3 The Monomorphism Restriction

The Haskell type system contains a restriction related to type classes that is not found in ordinary
Hindley-Milner type systems: the monomorphism restriction. The reason for this restriction is
related to a subtle type ambiguity and is explained in full detail in the Report (x4.5.5). A simpler
explanation follows:

The monomorphism restriction says that any identi�er bound by a pattern binding (which in-
cludes bindings to a single identi�er), and having no explicit type signature, must be monomorphic.
An identi�er is monomorphic if is either not overloaded, or is overloaded but is used in at most one
speci�c overloading and is not exported.

Violations of this restriction result in a static type error. The simplest way to avoid the problem
is to provide an explicit type signature. Note that any type signature will do (as long it is type
correct).

A common violation of the restriction happens with functions de�ned in a higher-order manner,
as in this de�nition of sum from the Standard Prelude:

sum = foldl (+) 0

As is, this would cause a static type error. We can �x the problem by adding the type signature:

sum :: (Num a) => [a] -> a

Also note that this problem would not have arisen if we had written:

sum xs = foldl (+) 0 xs

because the restriction only applies to pattern bindings.

13 Arrays

Ideally, arrays in a functional language would be regarded simply as functions from indices to values,
but pragmatically, in order to assure eÆcient access to array elements, we need to be sure we can
take advantage of the special properties of the domains of these functions, which are isomorphic
to �nite contiguous subsets of the integers. Haskell, therefore, does not treat arrays as general
functions with an application operation, but as abstract data types with a subscript operation.

Two main approaches to functional arrays may be discerned: incremental and monolithic def-
inition. In the incremental case, we have a function that produces an empty array of a given size
and another that takes an array, an index, and a value, producing a new array that di�ers from
the old one only at the given index. Obviously, a naive implementation of such an array semantics
would be intolerably ineÆcient, either requiring a new copy of an array for each incremental rede�-
nition, or taking linear time for array lookup; thus, serious attempts at using this approach employ
sophisticated static analysis and clever run-time devices to avoid excessive copying. The monolithic
approach, on the other hand, constructs an array all at once, without reference to intermediate
array values. Although Haskell has an incremental array update operator, the main thrust of the
array facility is monolithic.

Arrays are not part of the Standard Prelude|the standard library contains the array operators.
Any module using arrays must import the Array module.

13.1 Index types

The Ix library de�nes a type class of array indices:

class (Ord a) => Ix a where

range :: (a,a) -> [a]

index :: (a,a) a -> Int

inRange :: (a,a) -> a -> Bool

Instance declarations are provided for Int, Integer, Char, Bool, and tuples of Ix types up to length
5; in addition, instances may be automatically derived for enumerated and tuple types. We regard
the primitive types as vector indices, and tuples as indices of multidimensional rectangular arrays.
Note that the �rst argument of each of the operations of class Ix is a pair of indices; these are
typically the bounds (�rst and last indices) of an array. For example, the bounds of a 10-element,
zero-origin vector with Int indices would be (0,9), while a 100 by 100 1-origin matrix might have
the bounds ((1,1),(100,100)). (In many other languages, such bounds would be written in a

form like 1:100, 1:100, but the present form �ts the type system better, since each bound is of
the same type as a general index.)

The range operation takes a bounds pair and produces the list of indices lying between those
bounds, in index order. For example,

range (0,4) ) [0,1,2,3,4]

range ((0,0),(1,2)) ) [(0,0), (0,1), (0,2), (1,0), (1,1), (1,2)]

The inRange predicate determines whether an index lies between a given pair of bounds. (For a
tuple type, this test is performed component-wise.) Finally, the index operation allows a particular
element of an array to be addressed: given a bounds pair and an in-range index, the operation yields
the zero-origin ordinal of the index within the range; for example:

index (1,9) 2 ) 1

index ((0,0),(1,2)) (1,1) ) 4

13.2 Array Creation

Haskell's monolithic array creation function forms an array from a pair of bounds and a list of
index-value pairs (an association list):

array :: (Ix a) => (a,a) -> [(a,b)] -> Array a b

Here, for example, is a de�nition of an array of the squares of numbers from 1 to 100:

squares = array (1,100) [(i, i*i) | i <- [1..100]]

This array expression is typical in using a list comprehension for the association list; in fact, this
usage results in array expressions much like the array comprehensions of the language Id [6].

Array subscripting is performed with the in�x operator !, and the bounds of an array can be
extracted with the function bounds:

squares!7 ) 49

bounds squares ) (1,100)

We might generalize this example by parameterizing the bounds and the function to be applied to
each index:

mkArray :: (Ix a) => (a -> b) -> (a,a) -> Array a b

mkArray f bnds = array bnds [(i, f i) | i <- range bnds]

Thus, we could de�ne squares as mkArray (\i -> i * i) (1,100).

Many arrays are de�ned recursively; that is, with the values of some elements depending on the
values of others. Here, for example, we have a function returning an array of Fibonacci numbers:

fibs :: Int -> Array Int Int

fibs n = a where a = array (0,n) ([(0, 1), (1, 1)] ++

[(i, a!(i-2) + a!(i-1)) | i <- [2..n]])

Another example of such a recurrence is the n by n wavefront matrix, in which elements of the
�rst row and �rst column all have the value 1 and other elements are sums of their neighbors to
the west, northwest, and north:

wavefront :: Int -> Array (Int,Int) Int

wavefront n = a where

a = array ((1,1),(n,n))

([((1,j), 1) | j <- [1..n]] ++

[((i,1), 1) | i <- [2..n]] ++

[((i,j), a!(i,j-1) + a!(i-1,j-1) + a!(i-1,j))

| i <- [2..n], j <- [2..n]])

The wavefront matrix is so called because in a parallel implementation, the recurrence dictates that
the computation can begin with the �rst row and column in parallel and proceed as a wedge-shaped
wave, traveling from northwest to southeast. It is important to note, however, that no order of
computation is speci�ed by the association list.

In each of our examples so far, we have given a unique association for each index of the array
and only for the indices within the bounds of the array, and indeed, we must do this in general
for an array be fully de�ned. An association with an out-of-bounds index results in an error; if an
index is missing or appears more than once, however, there is no immediate error, but the value of
the array at that index is then unde�ned, so that subscripting the array with such an index yields
an error.

13.3 Accumulation

We can relax the restriction that an index appear at most once in the association list by specifying
how to combine multiple values associated with a single index; the result is called an accumulated
array:

accumArray :: (Ix a) -> (b -> c -> b) -> b -> (a,a) -> [Assoc a c] -> Array a b

The �rst argument of accumArray is the accumulating function, the second is an initial value (the
same for each element of the array), and the remaining arguments are bounds and an association
list, as with the array function. Typically, the accumulating function is (+), and the initial value,
zero; for example, this function takes a pair of bounds and a list of values (of an index type) and
yields a histogram; that is, a table of the number of occurrences of each value within the bounds:

hist :: (Ix a, Integral b) => (a,a) -> [a] -> Array a b

hist bnds is = accumArray (+) 0 bnds [(i, 1) | i <- is, inRange bnds i]

Suppose we have a collection of measurements on the interval [a; b), and we want to divide the
interval into decades and count the number of measurements within each:

decades :: (RealFrac a) => a -> a -> [a] -> Array Int Int

decades a b = hist (0,9) . map decade

where decade x = floor ((x - a) * s)

s = 10 / (b - a)

13.4 Incremental updates

In addition to the monolithic array creation functions, Haskell also has an incremental array update
function, written as the in�x operator //; the simplest case, an array a with element i updated to
v, is written a // [(i, v)]. The reason for the square brackets is that the left argument of (//)
is an association list, usually containing a proper subset of the indices of the array:

(//) :: (Ix a) => Array a b -> [(a,b)] -> Array a b

As with the array function, the indices in the association list must be unique for the values to be
de�ned. For example, here is a function to interchange two rows of a matrix:

swapRows :: (Ix a, Ix b, Enum b) => a -> a -> Array (a,b) c -> Array (a,b) c

swapRows i i' a = a // ([((i ,j), a!(i',j)) | j <- [jLo..jHi]] ++

[((i',j), a!(i ,j)) | j <- [jLo..jHi]])

where ((iLo,jLo),(iHi,jHi)) = bounds a

The concatenation here of two separate list comprehensions over the same list of j indices is,
however, a slight ineÆciency; it's like writing two loops where one will do in an imperative language.
Never fear, we can perform the equivalent of a loop fusion optimization in Haskell:

swapRows i i' a = a // [assoc | j <- [jLo..jHi],

assoc <- [((i ,j), a!(i',j)),

((i',j), a!(i, j))] ]

where ((iLo,jLo),(iHi,jHi)) = bounds a

13.5 An example: Matrix Multiplication

We complete our introduction to Haskell arrays with the familiar example of matrix multiplication,
taking advantage of overloading to de�ne a fairly general function. Since only multiplication and
addition on the element type of the matrices is involved, we get a function that multiplies matrices
of any numeric type unless we try hard not to. Additionally, if we are careful to apply only (!)

and the operations of Ix to indices, we get genericity over index types, and in fact, the four row
and column index types need not all be the same. For simplicity, however, we require that the left
column indices and right row indices be of the same type, and moreover, that the bounds be equal:

matMult :: (Ix a, Ix b, Ix c, Num d) =>

Array (a,b) d -> Array (b,c) d -> Array (a,c) d

matMult x y = array resultBounds

[((i,j), sum [x!(i,k) * y!(k,j) | k <- range (lj,uj)])

| i <- range (li,ui),

j <- range (lj',uj') ]

where ((li,lj),(ui,uj)) = bounds x

((li',lj'),(ui',uj')) = bounds y

resultBounds

| (lj,uj)==(li',ui') = ((li,lj'),(ui,uj'))

| otherwise = error "matMult: incompatible bounds"

As an aside, we can also de�ne matMult using accumArray, resulting in a presentation that more
closely resembles the usual formulation in an imperative language:

matMult x y = accumArray (+) 0 resultBounds

[((i,j), x!(i,k) * y!(k,j))

| i <- range (li,ui),

j <- range (lj',uj')

k <- range (lj,uj) ]

where ((li,lj),(ui,uj)) = bounds x

((li',lj'),(ui',uj')) = bounds y

resultBounds

| (lj,uj)==(li',ui') = ((li,lj'),(ui,uj'))

| otherwise = error "matMult: incompatible bounds"

We can generalize further by making the function higher-order, simply replacing sum and (*)

by functional parameters:

genMatMult :: (Ix a, Ix b, Ix c) =>

([f] -> g) -> (d -> e -> f) ->

Array (a,b) d -> Array (b,c) e -> Array (a,c) g

genMatMult sum' star x y =

array resultBounds

[((i,j), sum' [x!(i,k) `star` y!(k,j) | k <- range (lj,uj)])

| i <- range (li,ui),

j <- range (lj',uj') ]

where ((li,lj),(ui,uj)) = bounds x

((li',lj'),(ui',uj')) = bounds y

resultBounds

| (lj,uj)==(li',ui') = ((li,lj'),(ui,uj'))

| otherwise = error "matMult: incompatible bounds"

APL fans will recognize the usefulness of functions like the following:

genMatMult maximum (-)

genMatMult and (==)

With the �rst of these, the arguments are numeric matrices, and the (i ; j )-th element of the result
is the maximum di�erence between corresponding elements of the i-th row and j -th column of the
inputs. In the second case, the arguments are matrices of any equality type, and the result is a
Boolean matrix in which element (i ; j ) is True if and only if the i-th row of the �rst argument and
j -th column of the second are equal as vectors.

Notice that the element types of genMatMult need not be the same, but merely appropriate
for the function parameter star. We could generalize still further by dropping the requirement
that the �rst column index and second row index types be the same; clearly, two matrices could
be considered conformable as long as the lengths of the columns of the �rst and the rows of the
second are equal. The reader may wish to derive this still more general version. (Hint: Use the
index operation to determine the lengths.)

14 The Next Stage

A large collection of Haskell resources is available on the web at haskell.org. Here you will �nd
compilers, demos, papers, and much valuable information about Haskell and functional program-
ming. Haskell compilers or interpreters run on almost all hardware and operating systems. The
Hugs system is both small and portable { it is an excellent vehicle for learning Haskell.

